


ORIC



DES PROGRAMMES POUR VOTRE ORIC

MICHEL PIOT

CEDIC/FERNAND NATHAN

DES PROGRAMMES POUR VOTRE ORIC

Michel Piot

ÉDITIONS CÉDIC

Dans la même collection

Initiation au BASIC TO7 — *Christine et François-Marie Blondel.*

Le BASIC D.O.S. du TO7 — *Christine et François-Marie Blondel.*

Un ordinateur à la maison — *Jean Delcourt.*

Un ordinateur en fête — *Serge Pouts Lajus.*

Un ordinateur et des jeux — *Jean-Pascal Duclos.*

Premiers pas avec le ZX-SPECTRUM — *Ian Stewart & Robin Jones (Adaptation de Claire Touchard).*

Plus loin avec le ZX-SPECTRUM — *Ian Stewart & Robin Jones (Traduction de Claire Touchard).*

Guide pratique de l'ordinateur personnel IBM — *Dalloz - Emery - Portefaix - Boisgontier-Salzman*

LOGO, des ailes pour l'esprit — *Horacio C. Reggini (Traduction de E. et S. Letelier et A. Schake).*

Jeux vidéo, Jeux de demain — *Georges-Marie Becherraz et Alain Graber.*

Guide pratique de l'Oric - *Michel Bussac et Robert Lagoutte*

Ce volume porte la référence
ISBN 2-7124-0539-0

Toute reproduction, même partielle, de cet ouvrage est interdite. Une copie ou reproduction par quelque procédé que ce soit, photographie, photocopie, microfilm, bande magnétique, disque ou autre, constitue une contrefaçon passible des peines prévues par la loi du 11 mars 1975 sur la protection des droits d'auteur.

© CEDIC 1983

CEDIC, 32, boulevard Saint-Germain, 75005 - PARIS

Sommaire

Introduction	5
1 Bonjour	9
2 Cadavre exquis	14
3 Chapeau	19
4 Le compte est bon	22
5 Calendrier	26
6 Course	30
7 Dactyloric	34
8 La gaufre empoisonnée	37
9 Décodez	43
10 Héron	47
11 Loto	51
12 Mnémonique téléphonique.....	56
13 Mot mité	59
14 Que feriez-vous si... ..	63
15 Ration calorique	67
16 Rencontre	72
17 Rotation	76
18 Solitaire	80
19 Synthétiseur	85
20 Taquin	89
21 Télépathie	92
22 Test de Ruffier	97
23 Touche sensible	101
24 Z... mégalettres... A	104
Annexes	109
• Codes ASCII	
• Attributs	

Quels programmes choisir ?

Dans cet ouvrage, les programmes sont rangés par ordre alphabétique de leur titre. (Pourquoi pas ?). Pour vous y reconnaître un peu, nous vous conseillons les programmes suivants.

Si vous voulez

... amuser vos amis :

- | | | |
|------------------|---------------------------|-----------------|
| 1. Bonjour | 2. Cadavre exquis | 7. Dactyloric |
| 11. Loto | 14. Que feriez vous si... | 16. Rencontre |
| 19. Synthétiseur | 23. Touche sensible | 24. Mégalettres |

... faire semblant d'être sérieux :

- | | | |
|----------------|----------------------|---------------------|
| 3. Chapeau | 5. Calendrier | 8. Décodez |
| 12. Mnémonique | 15. Ration calorique | 22. Test de Ruffier |

... jouer le jeu avec un ordinateur :

- | | | |
|----------------------|--------------|-----------------------|
| 4. Le compte est bon | 6. Course | 9. Gaufre empoisonnée |
| 10. Héron | 13. Mot mité | 17. Rotation |
| 18. Solitaire | 20. Taquin | 21. Télépathie |

... et si vous désirez parcourir rapidement les possibilités de votre ORIC, voici une sélection spectaculaire et amusante :

- 1. Bonjour
- 3. Chapeau
- 7. Dactyloric
- 17. Rotation
- 19. Synthétiseur
- 24. Mégalettres.

Introduction

J'ai rassemblé dans ce livre 24 petits programmes qui ont tous été testés réellement dans la version correspondant exactement aux listings proposés et qui fonctionnent sur l'ORIC 16 K (et donc aussi sur l'ORIC 48 K).

Chacun d'eux propose un divertissement dans une présentation soignée (couleur, musique, etc.).

Au-delà de l'aspect récréatif non négligeable qu'ils offrent, ces programmes pourront vous apprendre, si vous ne le savez déjà et si vous en avez l'envie, comment on communique avec l'ordinateur ORIC.

Sauf en de rares exceptions, chaque ligne ne comporte qu'une seule instruction afin de faciliter l'étude du programme et les adaptations si vous désirez en faire.

Dans certains programmes vous pourrez extraire des sous-routines (écriture en double hauteur paramétrée, acquisition d'une réponse en un endroit quelconque de l'écran, exécution d'un air de musique par exemple) en vue de les inclure dans vos propres réalisations.

Quelques remarques

- Pour entrer en mémoire une ligne de programme affichée à l'écran, n'oubliez pas de frapper sur la touche marquée RETURN
- Si vous manquez de temps ou de courage lors de la frappe, vous pourrez ignorer les lignes commençant par REM sans que cela altère l'exécution.
- Prenez garde à ne pas confondre le chiffre 0 (il est barré) et la lettre O que, contrairement aux machines à écrire, on obtient à l'aide de touches différentes !.
- Dans certains messages écrits entre guillemets figurent des espaces ; comptez-les soigneusement car leur nombre est souvent critique.
- Il arrive que 2 guillemets se suivent sans rien contenir :
ex. IF KEY\$ = ""....
cela est normal et représente ce qu'on appelle la "chaîne de caractères vide".
- J'ai fait une utilisation intense de CHR\$ à la suite de PRINT ou de PLOT. Ils sont, pour la plupart destinés à la gestion de l'apparition des caractères sur l'écran (couleurs, taille, clignotement). Vous pourrez effectuer toutes modifications que vous jugerez utiles en changeant la valeur de l'argument des CHR\$ (voir Annexes 1 et 2).
- Il arrive que 2 lignes consécutives soient identiques ou presque ; il ne s'agit pas d'erreurs mais d'une certaine manière d'écrire en double hauteur.
- Quand vous avez tapé complètement un programme, prenez la précaution de le sauvegarder sur une cassette avant de l'exécuter. En effet, si une faute de frappe déclençait un "plantage" de la machine nécessitant qu'on l'éteigne, vous pourrez alors recharger le programme à partir de la cassette et le corriger.

• Pour exécuter un programme contenu dans l'ordinateur, il faut taper les 3 lettres majuscules

RUN

puis presser la touche de retour chariot marquée RETURN

- Par ailleurs au cours de leur exécution, de nombreux programmes vous demandent de taper quelque chose au clavier : Sauf dans le cas où on vous invite à “presser une touche”, il faudra terminer votre réponse en pressant la touche RETURN ; cela est indispensable pour “rendre la main à l’ordinateur”.

- J’ai essayé d’utiliser un maximum d’instructions et de techniques différentes. C’est ainsi que les “attributs” (formes, couleurs...) des caractères affichés sur l’écran peuvent être positionnés par :

POKE (Synthétiseur ligne 127)

ou PLOT (Mégalettres ligne 42)


ou PRINT (Course ligne 20)

1. Bonjour

Pour 1 à 77 personnes

Présentation :

BONJOUR est un bref programme de bienvenue destiné à l'accueil de vos convives, un accueil plein de bonne humeur, de couleur et de musique.



Remarques :

Les TAB(25) réalisent en fait des TAB(12) car un "bug" de l'interpréteur de mon ORIC fait que les 13 premières entrées de l'argument de TAB ne sont pas prises en compte.

Si votre ORIC ne présentait pas ce "bug", il faudrait donc remplacer 25 par 12.

Remarquez comment on déplace horizontalement un caractère sur l'écran : il suffit qu'il traîne derrière lui un espace (" "); cette technique est également appliquée dans le programme COURSE.

(Merci à Serge Pouts Lajus de l'idée de ce programme et de quelques autres que nous avons trouvés dans l'ouvrage UN ORDINATEUR EN FÊTE pour l'ordinateur Thomson T07 - Éditions CEDIC).

Programme :

```
1 REM  
2 REM
```

```

3 REM
4 REM ** BONJOUR **
5 REM -----
6 REM
7 REM
8 DIM L$(20)
10 CLS
12 PAPER 0
15 INK 6
20 PRINT CHR$(17);CHR$(20)
25 M$=" Tapez votre prenom"
30 PRINT CHR$(142);M$
32 PRINT CHR$(142);M$
33 PRINT:PRINT:PRINT
34 M$=" Puis pressez la touche 'RETURN'"
35 PRINT M$
36 PRINT:PRINT:PRINT
38 INPUT N$
39 N$=LEFT$(N$,17)
40 E=LEN(N$)
45 CLS
50 GOSUB 500:REM * INVERSION *
60 PRINT CHR$(17)
100 POKE 616,9
101 PRINT
102 K$=CHR$(142)+"B O N J O U R"
103 M$=K$+CHR$(136)+CHR$(144)+CHR$(134)
107 PRINT TAB(25);M$
108 PRINT TAB(25);M$
110 GOSUB 600:REM * CADRE *
119 REM
120 REM * ECRITURE DU NOM *
121 REM
130 FOR I=1 TO E
135 IF I=1 THEN L$(I)=CHR$(ASC(L$(I))-32)
140 FOR J=35 TO 2*I+1 STEP -1
143 K=J
145 IF K>7 THEN K=K-7:GOTO 145
150 PLOT J,15,CHR$(K)+L$(I)+CHR$(6)
155 CALL HFB10
156 WAIT 7

```

```

157 CALL#FAFA
158 WAIT 3
160 NEXT J
163 FOR J=1 TO 10
165 CALL#FAFA
166 WAIT 10
167 CALL#FB10
169 NEXT J
180 NEXT I
184 REM
185 REM * REMISE EN ORDRE *
186 REM
190 PLOT 10,9,CHR$(10)
195 PLOT 10,10,CHR$(10)
205 WAIT 100
210 PLOT 2*E,15," "
220 PLOT 2*E,17,L$(E-1)
221 ZAP
225 WAIT 100
230 PLOT 2*E+2,15," "
240 PLOT 2*E,15,L$(E)
241 ZAP
250 WAIT 100
260 PLOT 2*E+2,17,L$(E-1)
270 PLOT 2*E,17," "
271 ZAP
275 WAIT 100
280 PLOT 2*E+2,15,L$(E-1)
290 PLOT 2*E+2,17," "
291 ZAP
293 REM
294 REM * ROCK *
300 A$="3131353538383A3A41413A3A38383535"
310 B$="36363A3A414143434444434341413A3A"
320 C$="38383C3C434345454646454543433C3C"
330 M$=A$
340 GOSUB 800
350 M$=A$
360 GOSUB 800
370 M$=B$
380 GOSUB 800

```

```

390 M#=A$
400 GOSUB 800
410 M#=C$
420 GOSUB 800
430 M#="3131353538383A38"
440 GOSUB 800
450 MUSIC 1,3,11,0
460 PLAY 1,0,1,4000
470 FOR I=1 TO 8:PRINT:NEXT I
480 PRINT CHR$(17);CHR$(20)
490 PRINT CHR$(17)
492 END
494 REM
495 REM * INVERSION *
496 REM
500 FOR I=1 TO E
510 L$(I)=MID$(M$,I,1)
520 NEXT I
530 V#=L$(E-1):L$(E-1)=L$(E):L$(E)=V#
540 RETURN
589 REM
590 REM ** CADRE **
591 REM
600 FOR C=2 TO 38
610 PLOT C,7,CHR$(126)
620 NEXT C
630 FOR L=7 TO 18
640 PLOT 38,L,CHR$(126)
650 NEXT L
660 FOR C=38 TO 2 STEP -1
670 PLOT C,18,CHR$(126)
680 NEXT C
690 FOR L=18 TO 7 STEP -1
700 PLOT 2,L,CHR$(126)
710 NEXT L
720 RETURN
789 REM
790 REM * MUSIQUE *
791 REM
800 FOR I=1 TO LEN(M$)-1 STEP 2
810 O=VAL(MID$(M$,I,1))

```

```
820 N=VAL("#"+MID$(M$,I+1,1))
830 MUSIC 1,D,N,0
840 PLAY 1,0,3,400
845 J=INT(RND(1)*7+1)
850 PLOT 13,9,CHR$(J)
852 PLOT 12,9,CHR$(ABS(7-J)+16)
855 PLOT 13,10,CHR$(J)
856 PLOT 12,10,CHR$(ABS(7-J)+16)
857 WAIT 20
860 NEXT I
870 RETURN
```

2. Cadavre exquis

Nombre de personnes : 1 à 81

Présentation :

Comme son nom l'indique il s'agit d'une version ORICIENNE d'un jeu cher à André Breton et ses amis.

Un à un, 5 joueurs sont invités à taper un groupe de mots répondant à certaines caractéristiques grammaticales.

L'affichage de ces 5 mots à la queue leu leu produit les phrases dont se repaissaient les familiers du Café de Flore et dont l'exemple le plus connu est :

LE CADAVRE EXQUIS BOIRA DU VIN NOUVEAU.

A vous maintenant, un demi-siècle plus tard, de goûter aux joies du surréalisme.

Remarques :

Afin de limiter l'aspect "irréaliste", le genre (masculin-féminin) et le nombre (singulier-pluriel) des 2 substantifs sont imposés mais avec une probabilité d'apparition égale pour chacune des 4 combinaisons ; cela est obtenu dans les lignes 21 à 28 grâce à la fonction RND.

Les lignes 19 et 20 simulent un RAMDOMIZE, c'est-à-dire qu'elles "rendent le hasard encore plus aléatoire" ; on peut traduire leur effet par "tant que l'utilisateur ne presse pas une touche, l'ordinateur avance dans sa série des nombres aléatoires".

Programme :

```
1 REM
2 REM
3 REM
4 REM ** CADAVRE EXDUIS **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 1
16 S$=" "
17 M$="veuillez presser une touche "
18 PLOT 4,11,M$
19 A=RND(1)
20 IF KEY$="" THEN 19
21 G$="masculin "
22 IF RND(1)>.5 THEN G$="feminin "
23 I$="masculin "
24 IF RND(1)>.5 THEN I$="feminin "
25 H$="singulier"
26 IF RND(1)>.5 THEN H$="pluriel"
27 J$="singulier"
28 IF RND(1)>.5 THEN J$="pluriel"
29 CLS
30 REM * * * * *
31 PLOT 6,3,CHR$(10)+"veuillez taper"
32 PLOT 6,4,CHR$(10)+"veuillez taper"
33 L=7
34 M$=" un nom "
35 N$=G$+H$
36 F=36
37 GOSUB 500
40 L=9
43 M$="precede d'un determinant defini"
45 N$=""
46 F=36
47 GOSUB 500
50 M$="le chat ;les epiciers ;ma voisine"
```

```

60 PLOT 1,16,M$
65 PLOT 1,18,"mes oreilles par exemple."
68 INPUT A$
70 CLS
80 REM * * * * *
90 L=7
92 M$="adjectif "
95 N$=G$+H$
100 F=36
105 GOSUB 500
110 INPUT B$
120 CLS
125 REM * * * * *
130 L=7
132 M$="verbe (3eme pers. du "
135 N$=H$+" )"
140 F=36
145 GOSUB 500
147 PRINT
160 INPUT C$
170 CLS
175 REM * * * * *
180 L=7
183 M$=" un nom "
185 N$=I$+J$
186 F=36
187 GOSUB 500
189 REM * * * * *
190 L=9
193 M$=" precede d'un determinant"
195 N$=""
196 F=36
197 GOSUB 500
198 M$="la pomme ;une vipere ;du fromage"
200 PLOT 1,12,M$
203 M$="des avions ;de la poudre ..."
205 PLOT 1,13,M$
220 INPUT D$
230 CLS
240 REM * * * * *
250 L=7

```

```

252 M$="adjectif "
255 N$=I$+J$
260 F=36
265 GOSUB 500
270 INPUT E$
300 CLS
310 L=7
320 M$=" VOICI VOTRE 'cadavre exquis'"
325 N$=""
330 N$=""
340 F=36
350 GOSUB 500
360 GOSUB 600
470 POKE 616,20
472 PRINT
475 PRINTA$;S$;B$;S$;C$;S$;D$;S$;E$
480 END
485 REM
490 REM * MESSAGE *
495 REM
500 PLOT 6,3,CHR$(10)+"veuillez taper"
502 PLOT 6,4,CHR$(10)+"veuillez taper"
505 PLOT 1,L,10:PLOT 2,L,19
507 PLOT 3,L,4:PLOT 4,L,M$+N$
510 M=L+1
515 PLOT 1,M,10:PLOT 2,M,19
517 PLOT 3,M,4:PLOT 4,M,M$+N$
520 PLOT F,L,22
530 PLOT F,M,22
535 POKE 616,20
540 PRINT
550 RETURN
585 REM
590 REM * MUSIQUE *
595 REM
600 MU$="414141454545484848"
610 FOR I=1 TO LEN(MU$)-1 STEP 2
620 D=VAL(MID$(MU$,I,1))
630 N=VAL(MID$(MU$,I+1,1))
640 MUSIC 1,D,N,0
650 PLAY 1,0,1,4000

```

655 WAIT 30
660 NEXT I
670 MUSIC 1,5,1,0
680 PLAY 1,0,1,4000
690 WAIT 120
700 MUSIC 1,3,1,0
710 PLAY 1,0,1,2000
720 WAIT 60
730 RETURN


3. Chapeau

Présentation :

Il s'agit de la représentation sur l'écran d'une courbe mathématique.

Que cela ne vous rebute pas car ce n'est pas pour son aspect mathématique qu'elle figure ici mais pour la forme qu'elle suggère.

Un point non négligeable à signaler : la réalisation de l'image complète demande... un certain temps !



Remarques :

Le programme est constitué de 2 boucles imbriquées au sein desquelles sont effectués de nombreux calculs que les amateurs de mathématiques étudieront.

Afin d'accélérer le déroulement du programme, les calculs sont effectués dès que possible : calculer Y^3 au sein de la formule de la ligne 180 plutôt qu'en ligne 160 aurait conduit à refaire le même calcul 60 fois au lieu d'une !

De plus, calculer un carré par $Y * Y$ est beaucoup plus rapide que par Y^2 .

Suggestions :

Les nombres affichés dans les lignes de texte du bas de l'écran correspondent à certaines valeurs calculées. Ceux qu'elles n'intéressent pas pourront supprimer les lignes 195, 235 et 236.

Il est possible de modifier la densité des points et par conséquent la rapidité d'exécution en modifiant les valeurs de SY % et

SX % initialisées en ligne 30 et qui sont les pas des boucles de calcul sur les coordonnées.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** CHAPEAU **
5 REM   ---
6 REM
7 REM
8 REM
10 DIM D%(1,239)
20 P=PI/180
30 D%=-180:F%=180:SYZ=6:SZX=6
40 HIRE$
50 PAPER 6
60 INK 0
61 PRINT CHR$(17)
62 PRINT
65 PRINT "   PATIENCE . . ."
69 REM
70 REM * INITIALISATION DU TABLEAU *
71 REM
80 FOR L=0 TO 239
90 D%(0,L)=-1
100 D%(1,L)=-1
110 NEXT L
119 REM
120 REM * BOUCLES DE CALCUL DES POINTS *
121 REM
130 FOR Y=D% TO F% STEP SYZ
140 Y1=(16-Y/2)/3
150 Y2=300-Y/2
160 Y3=Y*Y
170 FOR X=D% TO F% STEP SZX
180 R=P*SDR(X*X+Y3)
190 Z=100*COS(R)-30*COS(3*R)
195 PRINT "X=";X;"Y=";Y;"Z=";INT(Z)
200 DX%=110+X/2+Y1
```

```

210 DYZ=(Y2-Z)/3
220 D0Z=DZ(0,DXZ)
230 D1Z=DZ(1,DXZ)
235 PRINT "DX=";DXZ;"DY=";DYZ
236 PRINT CHR$(11);CHR$(11);
240 IF DXZ<0 OR DXZ>239 THEN 280
250 IF D0Z=-1 THEN 340
260 IF DYZ<=D0Z THEN 470
270 IF DYZ>=D1Z THEN 520
280 NEXT X
290 NEXT Y
300 END
310 REM
320 REM * DECISIONS ET ACTUALISATION *
330 REM
340 IF DXZ=0 THEN 420
350 IF DZ(0,DXZ-1)=-1 THEN 420
360 IF DZ(0,DXZ+1)=-1 THEN 420
370 DZ(0,DXZ)=(DZ(0,DXZ-1)+DZ(0,DXZ+1))/2
380 DZ(1,DXZ)=(DZ(1,DXZ-1)+DZ(1,DXZ+1))/2
390 CURSET DXZ,DYZ,1
400 GOTO 280 ' retour a la boucle
410 REM * * * * *
420 DZ(0,DXZ)=DYZ
430 DZ(1,DXZ)=DYZ
440 CURSET DXZ,DYZ,1
450 GOTO 280 ' retour a la boucle
460 REM * * * * *
470 CURSET DXZ,DYZ,1
480 DZ(0,DXZ)=DYZ
490 IF D1Z=-1 THEN DZ(1,DXZ)=DYZ
500 GOTO 280 ' retour a la boucle
510 REM * * * * *
520 CURSET DXZ,DYZ,1
530 DZ(1,DXZ)=DYZ
540 IF D0Z=-1 THEN DZ(1,DXZ)=DYZ
550 GOTO 280 ' retour a la boucle

```


4. Le compte est bon

Pour 1 personne

Présentation :

Jeu fort connu des fidèles auditeurs d'une certaine "radio périphérique".

Il s'agit de découvrir un nombre entier choisi par l'ordinateur. Pour cela vous disposez de 10 essais au terme de chacun desquels ORIC vous indiquera si votre proposition est inférieure, supérieure ou... égale au nombre mystérieux.



Remarques :

Le tirage de nombres aléatoires par l'ordinateur n'est en fait absolument pas aléatoire. C'est pourquoi dans ce jeu comme dans beaucoup d'autres, vous voyez apparaître le message "pressez une touche"

En effet, pendant qu'il attend cette pression, l'ordinateur tire des nombres au hasard (lignes 100 et 110) et c'est le dernier nombre tiré avant que vous ne sollicitiez le clavier qui sera à découvrir. Ainsi le hasard est garanti !

Suggestion :

Vous pourrez modifier l'intervalle des nombres à découvrir en modifiant la ligne 120 mais en leur gardant la forme générale

$$120N = \text{INT}(A * \text{SUP}) + \text{MIN}$$

le plus petit nombre possible sera alors MIN

le plus grand sera SUP + MIN - 1

Exemple

$$120N = (\text{INT}(A * 9000) + 1000)$$

fourni des nombres allant de 1000 à 9999.

En remplaçant le 10 de la ligne 150 par un autre nombre, vous changerez le nombre d'essais ; ne soyez cependant pas trop gourmand car si l'affichage sort de l'écran par le bas, le jeu sera interrompu par un message d'erreur engendré non par le programme mais par le système.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** LE COMPTE EST BON **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
16 REM
17 REM * REGLE DU JEU *
18 REM
20 X=8:Y=3:AT=10:FD=17:AV=3:FL=22
22 NB=2
23 M$="LE COMPTE EST BON "
24 GOSUB 1000
50 X=0:Y=9:AT=10:FD=19:AV=1
52 NB=2
54 M$="l'ordinateur vient de choisir "
55 GOSUB 1000
60 Y=11
62 M$="un nombre entier inferieur a 1000 "
65 GOSUB 1000
70 X=7:Y=15
72 M$="VOUS AVEZ 10 ESSAIS "
74 GOSUB 1000
80 Y=17
```

```

82 M$="POUR LE DECOUVRIR. "
84 GOSUB 1000
90 Y=21:AT=14:FD=23:AV=1
92 M$="pressez une touche "
94 GOSUB 1000
100 A=RND(1)
110 IF KEY$="" THEN 100
120 N=INT(A*1000)
130 POKE 616,8
131 PRINT
132 REM
133 REM * EFFACEMENT PARTIEL DE L'ECRAN *
134 REM
135 FOR I=1 TO 17
137 PRINT CHR$(14)
139 NEXT I
150 X=0:Y=13:AT=10
152 Q$="ESSAI NUMERO "
154 REM
155 REM * CORPS DU JEU *
156 REM
158 FOR I=1 TO 10
160 U$=MID$(STR$(I),2)
165 IF I<10 THEN U$=" "+U$
170 M$=Q$+U$+CHR$(4)
175 GOSUB 1000
180 C=20:L=8+I
190 GOSUB 2000
200 ES=INT(VAL(R$))
210 IF ES<>N THEN 225
212 Y=19:FD=20:AV=3
214 M$="B R A V O "
215 GOSUB 1000
216 POKE 616,1
217 PRINT
218 END
219 REM
220 REM * L'ESSAI N'EST PAS GAGNANT *
221 REM
225 RP$=" petit "
230 IF ES>N THEN RP$=" grand "

```

```

240 D$="trop"+RP$
250 PLOT 28,8+I,D$
260 NEXT I
270 REM
272 REM * 10 ESSAIS INFRUCTUEUX *
274 REM
300 Y=19:FD=20:AV=3
305 M$="P E R D U "
310 GOSUB 1000
320 Y=23
325 M$="LE NOMBRE ETAIT "
330 M$=M$+MID$(STR$(N),2)+" "
340 GOSUB 1000
350 POKE 616,1
360 PRINT
370 END
989 REM
990 REM * AFFICHAGE
991 REM
1000 N$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1010 FOR NN=1 TO NB
1020 PLOT X,Y+NN-1,N$+M$+CHR$(FL)
1030 NEXT NN
1040 RETURN
2000 PLOT C,L,8
2010 POKE 616,L
2020 PRINT
2030 POKE 617,C+2
2040 INPUT R$
2050 RETURN

```

5. Calendrier

Nombre de personnes : 1 à 83

Présentation :

Sans doute vous êtes-vous déjà demandé quel était le nom du jour correspondant à une date donnée. Faute d'avoir sous la main tous les calendriers des années passées et à venir vous ne trouvez pas de réponse à votre question.

Ce programme ne vous donnera pas seulement le nom du jour mais le calendrier complet pour le mois de votre choix d'une année donnée.

Remarques :

Tout tourne autour de la formule de la ligne 70 qui détermine le nom du jour du 1^{er} du mois qui vous intéresse, les paramètres G et F étant ajustés par les lignes 52 à 65 en fonction du numéro du mois.

La ligne 96 ne présente aucun intérêt compte tenu de la fourchette imposée pour les millésimes (1901 à 2099) mais serait utile dans les cas où vous voudriez élargir cette fourchette (voir suggestions).

La ligne 94 a pour effet de brancher en 100 lorsque le nombre A n'est pas un multiple de 4 (année non bissextile donc).

L'ordinateur ne connaissant a priori ni les noms des jours, ni les noms des mois, ni le nombre de jours constituant chaque mois, les lignes de données 900 à 930 contiennent ces renseignements qui seront utilisés, en particulier, par la boucle 85 à 87.

Suggestions :

Modifier le programme afin d'élargir la fourchette des millésimes (attention aux chausse-trappes).

Écrire une routine de sortie sur imprimante des douze tableaux correspondant à un année complète.

Programme :

```
1 REM  
2 REM  
3 REM
```

```

4 REM ** CALENDRIER **
5 REM      -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
16 DIM J$(50),K$(7)
17 REM
18 REM * TABLEAU DES NOMS DES JOURS *
19 REM
20 FOR I=1 TO 7
21 READ K$(I)
22 NEXT I
23 REM
25 X=6:Y=3:AT=10:FD=19:AV=4:FL=22
26 NB=2
27 M$="C A L E N D R I E R  "
28 GOSUB 1000
30 X=0:Y=9
31 M$="QUELLE ANNEE (1901 a 2099) "
32 GOSUB 1000
34 C=31:L=10
35 GOSUB 2000
36 A=INT(VAL(R$))
37 IF A<1901 OR A>2099 THEN ZAP:GOTO 35
40 Y=15
41 M$="NUMERO DU MOIS (1 a 12)  "
42 GOSUB 1000
44 L=16
45 GOSUB 2000
46 M=INT(VAL(R$))
47 IF M<1 OR M>12 THEN ZAP:GOTO 45
48 REM
49 REM * APPLICATION DE LA FORMULE *
50 REM
52 G=A-1
55 IF M>2 THEN G=A
60 F=M+13
65 IF M>2 THEN F=M+1

```

```

70 Q=INT(365.25*G)+INT(30.6*F)-621048
75 N=INT(Q-INT(Q/7)*7)
77 IF N=0 THEN N=7
80 N=N-1
81 REM
82 REM * ACQUISITION DU NOM DU MOIS *
83 REM * ET DE SON NOMBRE DE JOURS. *
84 REM
85 FOR I=1 TO M
86 READ MO$,NJ
87 NEXT I
89 REM
90 REM * ANNEE BISSEXTILE ? *
91 REM
94 IF A/4<>INT(A/4) THEN 100
96 IF A=1900 THEN 100
98 IF M=2 THEN NJ=29
100 CLS
104 REM
105 REM * NOM DU MOIS ET MILLESIME *
106 REM
110 X=10:Y=3
120 M#=MO$+" "+MID$(STR$(A),2)+" "
130 GOSUB 1000
140 PLOT 1,8,1
149 REM
150 REM * NOMS DES JOURS *
151 REM
200 FOR I=1 TO 7
260 FOR I=1 TO N+NJ-1
265 I1=INT(I/7)
270 K#=J$(I)
280 IF VAL(K#)<10 THEN K#=" "+K#
290 PLOT 5+4*(I-I1*7),10+2*I1,K#
300 NEXT I
889 REM
890 REM * DONNEES FIXES *
891 REM
900 DATA Lu, Ma, Me, Je, Ve, Sa, Di
910 DATA JANVIER, 31, FEVRIER, 28, MARS, 31
920 DATA AVRIL, 30, MAI, 31, JUIN, 30, JUILLET

```

```

925 DATA 31,AOUT,31,SEPTEMBRE,30,OCTOBRE
930 DATA 31,NOVEMBRE,30,DECEMBRE,31
980 END
990 REM
992 REM * AFFICHAGE *
995 REM
1000 N$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1010 FOR NN=1 TO NB
1020 PLOT X,Y+NN-1,N$+M$+CHR$(FL)
1030 NEXT NN
1040 RETURN
1989 REM
1990 REM * POSITIONNE LE CURSEUR *
1991 REM
2000 PLOT C,L,8
2010 POKE 616,L
2020 PRINT
2030 POKE 617,C+2
2040 INPUT R$
2050 RETURN
210 PLOT 1+I*4,8,K$(I)
220 NEXT I
224 REM
225 REM ** CORPS DU CALENDRIER **
226 REM
230 FOR I=1 TO N+NJ
240 J$(I+N-1)=MID$(STR$(I),2)
250 NEXT I
254 REM
255 REM * AFFICHAGE DES QUANTIEMES *
256 REM

```

6. Course

Nombre de joueurs : 2 à 9 (plus les spectateurs)

Présentation :

Ce programme vous propose une compétition entre 2 à 9 joueurs ; il s'agit d'une course en ligne organisée sur une largeur d'écran entre des concurrents représentés par les 6 premières lettres de leur prénoms.

S'il y a un Michel et une Micheline parmi les coureurs il faudra qu'ils se surnomment MIC 1 et MIC 2 afin de ne pas créer d'ambiguïté.

Les concurrents avancent (ou restent sur place) chacun leur tour et du haut de l'écran vers le bas mais rassurez-vous, toutes les positions sur la ligne de départ se valent car il y a une gestion des "dead heads".

Pour apporter plus d'intérêt à la course, il vous faudra parier ou décider que le premier arrivé est le plus ceci ou le moins cela.

Remarques :

Afin de répartir régulièrement les lignes de course sur l'écran, la ligne 140 détermine l'écart E à respecter entre 2 lignes, le nombre E ainsi calculé étant exploité en ligne 170.

Pour vous convaincre de la nécessité des 2 espaces entre guillemets qui figurent en ligne 85, vous n'aurez qu'à les supprimer ! Si la course se déroule trop rapidement ou trop lentement à votre goût, vous pourrez modifier la temporisation de la ligne 295 ou la supprimer en écrivant 295 REM (on ne peut pas supprimer

purement et simplement la ligne 295 car, en 285 on trouve un GOTO 295 qui, sans cette précaution, déclencherait un "UN DEFINED STATEMENT ERROR" !).

Suggestion :

Faire un programme du même genre mais qui propose un parcours plus long (mais de même longueur pour tous !) ; ce pourrait être un (ou plusieurs) allers et retours ou un circuit en forme de 8 ou un "tour de stade" avec positions de départ décalées.

Programme :

```
1 REM
2 REM
3 REM
4 REM **      COURSE      **
5 REM      -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 0
15 INK 6
16 NT=400
17 PRINT CHR$(17)
18 M$="  LA COURSE VA BIENTOT COMMENCER"
20 PRINT CHR$(142)+CHR$(133)+M$
25 PRINT CHR$(142)+CHR$(133)+M$
27 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
28 M$="  LES CONCURRENTS SONT PRIES DE"
30 PRINT CHR$(138)+M$
35 PRINT CHR$(138)+M$
37 PRINT:PRINT
38 M$="  SE PRESENTER AU DEPART."
40 PRINT CHR$(138),M$
45 PRINT CHR$(138),M$
46 PRINT:PRINT:PRINT
48 REM
49 REM * ACQUISITION DU NOMBRE DE COUREURS *
50 REM
51 M$="Combien de coureurs (max=9)"
52 PRINT CHR$(132);M$
53 PRINT:PRINT:PRINT
55 INPUT N
56 IF N<10 AND N>1 THEN 64
57 PRINT CHR$(11);CHR$(11)
58 PING
59 GOTO 55
60 REM
61 REM *EFFACEMENT PARTIEL DE L'ECRAN *
62 REM
```

```

64 FOR L=5 TO 24
65 POKE 616,L
66 PRINT CHR$(14)
67 NEXT L
68 POKE 616,10:PRINT
69 REM
70 REM * ACQUISITION ET NORMALISATION DES NOMS *
71 REM
74 FOR I=1 TO N
75 PRINT " Nom du coureur No ";I;
80 INPUT N$
82 IF LEN(N$)<6 THEN N$=N$+" ":GOTO 82
85 A$(I)=LEFT$(N$,6)+" "
90 X(I)=32
100 NEXT I
105 CLS
110 REM
120 REM *** LIGNE DE DEPART ***
130 REM
135 K$="^....."
140 E=20/(N-1)
150 FOR L=1 TO N
155 C=L
157 IF C>7 THEN C=C-7
160 C$=K$+A$(L)
170 PLOT 1,2+INT((L-1)*E),CHR$(C)+C$
180 NEXT L
190 REM
200 REM ** LA COURSE **
205 REM
207 M$="presser une touche pour commencer"
210 PLOT 2,26,M$
220 IF KEY$="" THEN H=RND(1):GOTO 220
225 M$=" C ' E S T P A R T I ! ! "
230 PLOT 2,26,M$
234 REM
235 REM * FAIT AVANCER DU PAS *
236 REM * UNE FOIS CHAQUE COUREUR *
237 REM
240 FOR I=1 TO N
242 SOUND 1,NT,8

```

```
245 IF V(I)<>0 THEN 300
250 IF RND(1)>.5 THEN 300
270 X(I)=X(I)-1
280 PLOT X(I),2+(I-1)*E,A$(I)
285 IF X(I)<>3 THEN 295
290 PLOT 0,2+(I-1)*E/CHR$(V+49)
291 PING
292 FI=FI+1:V(I)=1:W=W+1
295 WAIT V(I)+10-N
300 NEXT I
305 IF W<>0 THEN V=V+W:W=0
307 NT=NT-4
310 IF FI<N THEN 240
320 POKE 616,25
330 PRINT CHR$(17);
340 END
```

7. Dactyloric

Nombre de personnes : 1


Présentation :

Faire que vous connaissiez mieux la position des lettres sur le clavier de votre ORIC, voilà le but de ce programme.

L'ordinateur affiche en double hauteur chacune des 26 lettres de l'alphabet en désordre et vous disposez d'un délai plus ou moins long selon le niveau de difficulté choisi pour taper la lettre proposée ; un PING ! indique que vous avez réussi dans le délai imparti.

Après la 26^e lettre, votre note apparaît.

A la demande de niveau, répondre 1 pour débutant jusqu'à 9 pour expert.



Remarques :

Les codes ASCII des lettres qui apparaissent à l'écran sont contenus dans un tableau T qui est rempli en 2 temps :

1) les 26 codes sont écrits dans les 26 cases du tableau et dans l'ordre "alphabétique" (lignes 17 à 20)

2) afin de mettre un peu de désordre, le programme effectue 20 permutations de contenu entre 2 cases (lignes 32 à 36).

La boucle 510 à 550 produit l'effet de "journal lumineux" sur la suite des lettres.

Suggestions :

Vous pourrez améliorer le programme afin qu'il propose également les chiffres, les signes de ponctuation et les caractères non

littéraux (remarquez que le 64 qui figure dans la ligne 19 est le code ASCII de la lettre A).

Si les délais ne vous conviennent pas, vous pourrez les ajuster à votre guise en modifiant la ligne 560.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** DACTYLORIC **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 0
17 DIM T(26)
18 FOR I=1 TO 26
19 T(I)=I+64
20 NEXT I
22 X=2:Y=9:AT=10:FD=21:AV=3:FL=22
23 NB=2:M$="  Pressez une touche  "
24 GOSUB 1000
25 Y=11
26 NB=2:M$="puis patientez un instant "
27 GOSUB 1000
28 A=RND(1)
29 IF KEY$="" THEN 28
30 PING
32 FOR I=1 TO 20
33 R=INT(RND(1)*26+1)
34 S=INT(RND(1)*26+1)
35 V=T(R):T(R)=T(S):T(S)=V
36 NEXT I
37 FOR I=1 TO 26
38 A$=A$+CHR$(T(I))
39 NEXT I
40 B$="
45 C$=B$+A$
```

```

60 CLS
70 X=7:Y=3:AT=10:FD=19:AV=4:FL=22
71 NB=2:M$="D A C T Y L O R I C "
72 GOSUB 1000
80 X=5:Y=9:FD=21:AV=3
81 NB=2:M$="NIVEAU (1 a 9) : "
82 GOSUB 1000
85 GET A$
86 A=VAL(A$)
87 IF A<1 OR A>9 THEN ZAP:GOTO 80
90 PLOT 27,9,CHR$(1)+A$
91 PLOT 27,10,CHR$(1)+A$
99 REM
100 REM * DEFILEMENT *
110 REM
500 FOR Z=1 TO 26
510 FOR W=1 TO 26
520 V$=MID$(C$,Z+W,1)
530 PLOT W+4,14,V$
550 NEXT W
555 PLOT 17,17,CHR$(10)+V$
557 PLOT 17,18,CHR$(10)+V$
560 WAIT 200-20*A
564 IF V$=KEY$ THEN SC=SC+1:PING
580 NEXT Z
590 X=1:Y=21:FD=23:AV=1
600 M$=MID$(STR$(SC),2)+" sur 26 "
610 GOSUB 1000
620 END
985 REM
990 REM * MESSAGE *
995 REM
1000 N$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1005 FOR NN=1 TO NB
1010 PLOT X,Y+NN-1,N$+M$+CHR$(FL)
1020 NEXT NN
1030 RETURN

```

8. La gaufre empoisonnée

Pour 2 personnes

Présentation :


Une appétissante gaufre est proposée à votre appétit, dommage qu'elle contienne un poison dans son coin inférieur gauche.

Chaque joueur à son tour doit croquer dans la gaufre mais malheur à celui qui se trouvera dans l'obligation d'avaler le poison.

Le choix de chaque portion est laissé au gré des joueurs qui en décident en déplaçant le "pavé rouge" à l'aide de 4 touches qui portent les flèches.

En appuyant sur la barre d'espace le joueur engloutit ce qui reste de la gaufre dans le rectangle ayant le pavé rouge pour coin inférieur gauche.

Malheureux celui dont c'est le tour alors que le poison clignote !



Remarques :

Malgré les diverses tailles qu'elle peut avoir, la gaufre reste toujours bien centrée dans la partie libre de l'écran.

Cela est obtenu grâce aux calculs des marges gauche et supérieure qui figurent aux lignes 72 et 74.

Les limites concernant les dimensions étant imposées, le programme doit vérifier qu'elles sont respectées par le joueur ; c'est le rôle des lignes 60 et 70.

Dans ce programme, de nombreux tests se font directement sur l'écran à l'aide de l'instruction SCRN.

Suggestions :

Changer la position initiale du poison.

Modifier le programme avec le poison dans le coin supérieur gauche est facile.

Déposer le poison dans l'un des 2 coins droits posera des problèmes au moment où il faudra faire disparaître la bouchée de l'écran.

Plus délicates seront les modifications qui permettront de jouer avec le poison situé en un endroit quelconque de la gaufre, car il faut aussi changer la définition de la partie engloutie lorsqu'on mange le pavé rouge.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** GAUFRE EMPOISONNEE **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
20 LO=7
25 LA=5
27 C$=CHR$(126)
30 X=6:Y=1:AT=10:FD=17:AV=0:FL=22
35 NB=2
40 M$="LA GAUFRE EMPOISONNEE "
45 GOSUB 1000
50 X=2:Y=5:AV=3
51 M$="TAILLE DE LA GAUFRE "
52 GOSUB 1000
54 Y=9:FD=22:AV=4
55 M$="LONGUEUR (2 à 30) "
56 GOSUB 1000
57 C=30:L=10
58 GOSUB 2000
59 LO=INT(VAL(R$))-1
```

```

60 IF LO<1 OR LO>29 THEN ZAP:GOTO 58
64 Y=13:FD=22:AV=4
65 M$="LARGEUR (2 a 18) "
66 GOSUB 1000
67 C=30:L=14
68 GOSUB 2000
69 LA=INT(VAL(R$))-1
70 IF LA<1 OR LA>17 THEN ZAP:GOTO 68
72 MG=INT((38-LO)/2)
74 MH=INT((18-LA)/2+4)
75 CLS
76 POKE 616,23
78 PRINT CHR$(17)
80 X=6:Y=1:AT=10:FD=17:AV=0:FL=22
85 NB=2
90 M$="LA GAUFRE EMPOISONNEE "
95 GOSUB 1000
105 FOR L=0 TO LA
110 FOR C=0 TO LO
120 PLOT MG+C,MH+L,C$
130 NEXT C
140 NEXT L
150 PLOT MG,MH+L-1,CHR$(224)
160 Y=23:AT=12
170 NB=1
180 M$="pressez une touche "
190 GOSUB 1000
200 A=RND(1)
210 IF KEY$="" THEN 200
220 C=INT(RND(1)*LO+1)
230 L=INT(RND(1)*LA+1)
240 IF C=0 AND L=LA THEN 220
260 PLOT MG+C,MH+L,254
300 X=0:AT=8
310 M$="deplacez vous avec les 4 fleches "
320 GOSUB 1000
330 Y=24
340 M$="croquez avec la barre d'espaces. "
350 GOSUB 1000
400 GET A$
410 IF FI=1 THEN 980

```

```

415 A=ASC(A#)-7
420 IF A=25 THEN 600
430 IF A<1 OR A>4 THEN SHOOT:GOTO 400
440 ON A GOTO 450,460,470,480
450 CP=C-1
451 LP=L
452 GOTO 490
460 CP=C+1
461 LP=L
462 GOTO 490
470 LP=L+1
471 CP=C
472 GOTO 490
480 LP=L-1
481 CP=C
490 SC=SCRN(CP+MG,LP+MH)
492 IF SC=224 THEN 900
495 IF SC<>126 THEN SHOOT:GOTO 400
500 PLOT C+MG,L+MH,126
510 C=CP
520 L=LP
530 PLOT C+MG,L+MH,254
590 GOTO 400
592 REM
593 REM * ON CROQUE *
594 REM
600 ZAP
610 FOR I=C TO LD
620 PLOT I+MG,L+MH,6
630 NEXT I
635 FOR I=L TO 0 STEP -1
640 PLOT C+MG,I+MH,6
645 NEXT I
650 IF SCRN(C+MG-1,L+MH)=126 THEN 700
660 IF SCRN(C+MG,L+MH+1)=126 THEN 710
670 IF SCRN(C+MG-1,L+MH)=224 THEN 720
680 IF SCRN(C+MG,L+MH+1)=224 THEN 730
700 C=C-1:GOTO 800
710 L=L+1:GOTO 800
720 C=C-1:L=L-1:GOTO 740
730 C=C+1:L=L+1

```

```

740 IF SCRN(C+MG,L+MH)=126 THEN 800
750 FI=1
760 PLOT MG-1,LA+MH,12
765 PING
770 GOTO 400
800 PLOT C+MG,L+MH,254
805 PING
810 GOTO 400
900 IF SCRN(MG,LA+MH-1)=126 THEN 950
910 IF SCRN(MG+1,LA+MH)=126 THEN 960
920 GOTO 400
950 PLOT C+MG,L+MH,126
952 C=C-1
954 L=L-1
956 GOTO 970
960 PLOT C+MG,L+MH,126
962 C=C+1
964 L=L+1
970 PLOT C+MG,L+MH,254
975 GOTO 400
980 M$="65635133"
985 GOSUB 3000
987 PRINT CHR$(17)
990 END
995 REM * AFFICHAGE *
996 REM
1000 M$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1010 FOR NN=1 TO NB
1020 PLOT X,Y+NN-1,N$+M$+CHR$(FL)
1030 NEXT NN
1040 RETURN
2000 PLOT C,L,8
2010 POKE 616,L
2020 PRINT
2030 POKE 617,C+2
2040 INPUT R$
2050 RETURN
3000 FOR I=1 TO LEN(M$)
3010 N=VAL("#"+MID$(M$,I,1))
3020 MUSIC 1,3,N,0
3030 PLAY 1,0,1,4000

```

3040 WAIT 100
3050 NEXT I
3060 RETURN

9. Décodez

Nombre de joueurs : 2

Présentation :


Par l'entreprise de l'ordinateur, les Services du Chiffre vous proposent un message parfaitement incompréhensible mais d'une importance capitale. De vos dons de décrypteur dépend peut être l'avenir de quelque chose !

Rassurez-vous, le codage est un simple décalage des lettres d'un certain nombre de "crans" dans l'alphabet. Par exemple, si le décalage est fixé à 4, tous les A du message deviendront des E, les B deviendront des F etc. et le message original ARBRE vous sera proposé sous la forme EVFVI.

Le message à déchiffrer étant sur l'écran, vous pouvez :

- soit presser une lettre
- soit déplacer l'index (flèche) en pressant l'une des touches marquées de flèches horizontales (→ et ←).

Si vous pressez une touche, un décodage qui respecte le choix que vous venez de faire pour remplacer la lettre fléchée sera réalisé ; si la proposition fournit une forme décodée conforme au message original, vous aurez gagné sinon, vous devrez faire une nouvelle proposition éventuellement précédée d'un déplacement de l'index.



Remarques :

La routine de codage ne modifie que les lettres majuscules (test de la ligne 650) et ceci afin, en particulier, de conserver les espa-

ces et d'avoir dans la phrase codée autant de mots que dans l'original.

Décaler la lettre Z de 4 ferait sortir de l'alphabet, il faut donc effectuer un décalage "modulo 26" et c'est le rôle de la ligne 675 grâce à laquelle Z sera codé D après décalage de 4.

Prétez une attention particulière à la gestion du déplacement de la flèche index (lignes 180 et 190). Pourquoi la flèche (que l'on obtient en tapant l'accent circonflexe ^) est-elle encadrée par 2 espaces (ligne 160) ? Remplacez " ^ " par " ^ " et vous comprendrez.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** DECODEZ **
5 REM      ----
6 REM
7 REM
8 REM
10 CLS
11 CLEAR
12 PAPER 4
15 INK 3
20 DIM A$(37),B$(37)
22 X=1:Y=5:AT=10:FD=17:AV=0:FL=20
23 M$="UN SEUL JOUEUR REGARDE L'ECRAN "
24 NB=2:GOSUB 1000
30 X=1:Y=11:AT=10:FD=19:AV=1:FL=20
32 M$="TAPEZ UN MESSAGE (1 à 38 car.) "
34 NB=2:GOSUB 1000
40 POKE 616,14
41 PRINT
42 INPUT A$
45 LM=LEN(A$)
46 IF LM>38 THEN ZAP:GOTO 10
50 X=1:Y=17
52 M$="CHOISISSEZ UN DECALAGE (1 à 25) "
54 NB=2:GOSUB 1000
56 POKE 616,20
```

```

57 PRINT:PRINT CHR$(14);
58 INPUT N
59 N=INT(N)
60 IF N<1 OR N>25 THEN ZAP:GOTO 50
69 REM
70 REM * ANALYSE *
71 REM
80 GOSUB 500
90 CLS
100 X=6:Y=5:AT=10:FD=19:AV=1:FL=20
102 M$="DECODEZ-CE MESSAGE "
104 NB=2:GOSUB 1000
110 GOSUB 600:REM * CODAGE *
120 XF=0
130 YF=13
140 PDKE 616,14
150 PRINT
153 REM
154 REM * ATTENTE D'UN PROPOSITION *
155 REM
160 PLOT XF,YF," ^ "
170 GET P$
180 IF P$=CHR$(9) AND XF<LM-1 THEN XF=XF+1
190 IF P$=CHR$(8) AND XF>0 THEN XF=XF-1
200 IF P$<"A" OR P$>"Z" THEN 400
210 IF A$(XF)<"A" OR A$(XF)>"Z" THEN 400
220 PLOT XF+1,YF+1,P$
230 WAIT 100
240 DC=ASC(P$)-ASC(A$(XF))
250 IF DC<0 THEN DC=DC+26
260 FOR Z=1 TO LM
270 C$=CHR$(ASC(A$(Z-1))+DC)
280 PLOT Z,14,C$
290 NEXT Z
300 IF P$=A$(XF) THEN PING:END
310 WAIT 200
320 PRINT CHR$(14);
330 ZAP
400 GOTO 160
480 END
485 REM

```

```

490 REM * ANALYSE DU MESSAGE *
495 REM
500 FOR Z=1 TO LM
520 A$(Z-1)=MID$(A$,Z,1)
530 NEXT Z
550 RETURN
585 REM
590 REM * CODAGE *
595 REM
600 FOR Z=1 TO LM
630 K=ASC(A$(Z-1))
640 REM * TEST-LETTRE *
650 IF K<65 OR K>90 THEN 690
655 REM
660 REM * DECALAGE *
665 REM
670 K=K+N
675 IF K>90 THEN K=K-26
680 B$(Z-1)=CHR$(K)
690 NEXT Z
700 PLOT 0,11,10
710 PLOT 0,12,10
720 FOR Z=1 TO LM
730 PLOT Z,11,B$(Z-1)
740 PLOT Z,12,B$(Z-1)
750 NEXT Z
760 RETURN
985 REM
990 REM * AFFICHAGE *
995 REM
1000 D$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1005 FOR NN=1 TO NB
1010 PLOT X,Y+NN-1,D$+M$+CHR$(FL)
1020 NEXT NN
1030 RETURN

```

10. Le héron

Nombre de joueurs : 1 ou plus

Présentation :

Vous souvenez-vous de cet échassier qui pour avoir négligé des mets convenables en espérant trouver mieux plus tard dû se contenter d'une portion congrue ?

Dans ce jeu, vous serez le héron à la recherche du plus grand nombre d'une série de 9.

Est-il préférable de choisir 873 alors qu'il figure en 6^e position ou faut-il espérer trouver mieux parmi les 3 nombres à venir ? La décision vous appartient.

L'ordinateur comptabilise les succès consécutifs sous forme d'une ligne d'étoiles bleues.

Un bon but à se fixer est d'aligner 5 de ces étoiles .

Si vous êtes 2 (ou plus), vous pourrez décider que le but est de faire mieux que le maximum d'étoiles obtenu précédemment.



Remarques :

Le choix des 9 nombres est confié à la boucle des lignes 190 à 196 qui les mémorise dans le tableau A et qui mémorise le plus grand d'entre eux dans la variable SUP.

L'écriture des nombres sur 2 niveaux est obtenu aux lignes 290 à 320 ; les 2 premières recopient le nombre au niveau inférieur, les 2 suivantes l'effacent du niveau supérieur.

Pour découvrir l'effet des lignes 655 et 656, vous n'aurez qu'à les supprimer purement et simplement.

La mémorisation du nombre de succès consécutifs se fait sous

forme d'une chaîne d'étoiles NF\$ à laquelle la ligne 657 ajoute une étoile en cas de réussite mais qui est impitoyablement vidée par la ligne 669 en cas d'échec.

Suggestion :

Améliorez le programme afin que, lorsqu'on joue à plusieurs, le score maximum réalisé soit constamment affiché en regard du nom du joueur qui l'a réussi le premier.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** LE HERON **
5 REM      ----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
18 M$="PRESSEZ UNE TOUCHE POUR COMMENCER"
19 PRINT CHR$(17)
20 PLOT 1,11,CHR$(14)+M$
21 PLOT 1,12,CHR$(14)+M$
30 X=RND(1)
35 K$=KEY$
40 IF K$="" THEN 30
50 CLS
95 M$="L'ordinateur vient de choisir"
100 PLOT 1,5,CHR$(10)+M$
110 PLOT 1,6,CHR$(10)+M$
115 M$="9 nombres entiers entre 10 et 999."
120 PLOT 1,7,CHR$(10)+M$
130 PLOT 1,8,CHR$(10)+M$
135 M$="Il va les dévoiler un a un."
140 PLOT 1,11,CHR$(10)+M$
150 PLOT 1,12,CHR$(10)+M$
155 M$="A vous de choisir le plus grand."
```

```

160 PLOT 1,15,CHR$(10)+M$
170 PLOT 1,16,CHR$(10)+M$
175 M$="pressez une touche"
180 PLOT 1,25,CHR$(12)+CHR$(1)+M$
185 GET K$
186 SUP=0
187 IF K$="F" THEN PRINT CHR$(17):END
190 FOR I=0 TO 8
192 A(I)=INT(RND(1)*989)+10
194 IF A(I)>SUP THEN SUP=A(I):J=I
196 NEXT I
198 CLS
200 PLOT 1,1,NF$
201 M$=" L E   H E R O N "
202 M$=CHR$(34)+M$+CHR$(34)
203 PLOT 8,3,CHR$(10)+CHR$(0)+M$
205 PLOT 8,4,CHR$(10)+CHR$(0)+M$
208 PLOT 1,9,CHR$(0)+CHR$(10)
210 PLOT 1,10,CHR$(0)+CHR$(10)
215 PLOT 1,11,10
220 PLOT 1,12,10
230 FOR I=0 TO 8
240 N$=MID$(STR$(A(I)),2)
250 IF LEN(N$)<3 THEN N$="0"+N$:GOTO 250
260 PLOT 4+I*4,9,N$
270 PLOT 4+I*4,10,N$
275 PLOT 4+I*4,11,CHR$(0)+"^"+CHR$(4)
277 PLOT 4+I*4,12,CHR$(0)+"^"+CHR$(4)
278 M$="choisissez-vous le nombre fleche ?"
279 PLOT 3,15,M$
280 M$="(0 ou N) "+CHR$(22)
281 PLOT 14,17,CHR$(0)+CHR$(23)+M$
282 POKE 616,19
283 PRINT CHR$(14);
284 GET R$
285 IF R$="O" THEN K=I:I=8:GOTO 600
290 PLOT 4+I*4,11,N$
300 PLOT 4+I*4,12,N$
310 PLOT 4+I*4,9,"   "
320 PLOT 4+I*4,10,"   "
600 NEXT I

```

```

605 IF R$<>"0" THEN 660
607 IF K=8 THEN 650
610 FOR L=K+1 TO 8
615 N$=MID$(STR$(A(L)),2)
617 IF LEN(N$)<3 THEN N$="0"+N$:GOTO 617
620 PLOT 4+L*4,11,N$
630 PLOT 4+L*4,12,N$
640 NEXT L
650 IF ABS(A(K)-SUP)>.1 THEN 660
652 PLOT 15,19,CHR$(14)+CHR$(1)+"BRAVO"
653 PLOT 15,20,CHR$(14)+CHR$(1)+"BRAVO"
654 PING
655 PLOT 3+J*4,9,1
656 PLOT 3+J*4,10,1
657 NF$=NF$+"*"
659 GOTO 670
660 M$="RATE ! LE PLUS GRAND ETAIT "
661 EXPLDDE
662 PLOT 2,19,CHR$(1)+M$+MID$(STR$(SUP),2)
663 IF J=8 THEN 666
664 PLOT 7+J*4,11,4
665 PLOT 7+J*4,12,4
666 PLOT 3+J*4,11,1
667 PLOT 3+J*4,12,1
668 PLOT 2,9," "
669 NF$=""
670 FOR I=22 TO 25
680 PLOT 1,I,20
685 PLOT 37,I,22
690 NEXT I
700 M$="PRESSEZ 'C' POUR RECOMMENCER"
710 PLOT 3,23,CHR$(6)+M$
720 M$="OU 'F' POUR CESSER CE JEU."
730 PLOT 3,25,CHR$(6)+M$
740 GOTO 185

```

11. Loto

Nombre de personnes : 1


Présentation :

Comment diable choisir les nombres à inclure dans sa grille de loto hebdomadaire !

Des spots publicitaires vous proposent de temps à autres des idées originales pour cela à la télévision.

Pour notre part nous laisserons ce choix aux bons soins de l'ordinateur qui, n'en doutez pas, vous donnera un jour le loto gagnant après avoir analysé votre prénom, votre date de naissance et la conjoncture astrale.

En prime vous aurez droit à la détermination de votre chiffre caractéristique : celui qui décide de tout concernant votre destin.



Remarques :

Les lignes 26 à 29 construisent les chaînes d'attributs de caractères W\$, X\$, Y\$ et Z\$ qui seront utilisées dans la suite du programme pour déclarer la taille, la couleur, l'apparence des messages.

Pour les décrypter, référez-vous à l'annexe qui traite des attributs-série.

Chacun des nombres constituant la combinaison est déterminé par la conjonction des lignes 850 et 870.

Le rôle de la boucle 890 à 910 est d'éviter la répétition d'un nombre dans une combinaison. Pour cela, un indicateur (DJ) est initialisé à 0 puis mis à 1 dans le seul cas où le nombre qui vient d'être tiré figure déjà parmi ceux qui le précèdent.

Suggestions :

Pour les amateurs de grilles multiples, vous pourrez modifier le programme afin qu'il détermine 2, 3... combinaisons de suite. Modifiez également la constitution des chaînes W\$, X\$, Y\$, Z\$ en remplaçant les nombres entre parenthèse par d'autres. Modifiez la couleur de fond, autrement dit : remplacer en ligne 12 le 0 de PAPER 0 par un autre nombre ; apparaît alors une traînée noire derrière les messages ; vous pourrez en changer la couleur en remplaçant les 144 (lignes 26, 32, 33, 50, 51, 125, 126) par le nombre approprié ; avec un peu de chance ou de réflexion vous la ferez "disparaître" en lui donnant la même couleur que PAPER.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** LOTO GAGNANT **
5 REM      -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 0
15 INK 3
17 PRINT
20 REM
21 REM * ATTRIBUTS DE CARACTERES *
22 REM
26 W$=CHR$(130)+CHR$(144)
27 X$=CHR$(142)+CHR$(145)+CHR$(131)
28 Z$=CHR$(150)+CHR$(132)
29 Y$=CHR$(138)+CHR$(147)+CHR$(129)
30 M$="  VOICI LE LOTO GAGNANT  "
32 PRINT SPC(5);X$;M$;CHR$(144)
33 PRINT SPC(5);X$;M$;CHR$(144)
35 PRINT:PRINT:PRINT
40 PRINT Z$;"TAPEZ VOTRE PRENDM";W$
42 PRINT:PRINT
45 INPUT P$
```

```

47 PRINT:PRINT:PRINT
48 M$="VOTRE DATE DE NAISSANCE : "
50 PRINT Y$;M$;CHR$(144)
51 PRINT Y$;M$;CHR$(144)
53 PRINT
55 PRINT Z$;"JOUR (de 1 a 31)      ";W$;
60 INPUT J
62 PRINT
65 PRINT Z$;"MOIS (de 1 a 12)     ";W$;
70 INPUT M
72 PRINT
75 PRINT Z$;"ANNEE (4 chiffres)   ";W$;
80 INPUT A
82 PRINT
90 REM
100 REM * CALCUL DU CHIFFRE *
110 REM
115 GOSUB 500
120 CLS
122 PRINT:PRINT:PRINT
125 PRINT Y$;"Votre chiffre est : ";K;CHR$(144)
126 PRINT Y$;"Votre chiffre est : ";K;CHR$(144)
160 REM
170 REM * MUSIQUE *
180 REM
185 GOSUB 600
190 PRINT:PRINT:PRINT
192 M$="VOICI VOTRE COMBINAISON GAGNANTE"
193 PRINT CHR$(134);M$
194 REM
195 REM * COMBINAISON *
196 REM
200 GOSUB 800
230 END
485 REM
490 REM ** CALCUL DU CHIFFRE **
495 REM
500 FOR I=1 TO LEN(P$)
530 L=ASC(MID$(P$,I,1))+L-64
540 NEXT I
550 L=L+J+M+A

```

```

560 K=L-INT(L/9)*9
570 IF K=0 THEN K=9
580 RETURN
585 REM
590 REM * MUSIQUE *
595 REM
600 PRINT:PRINT
605 M$="DE LA MUSIQUE PENDANT QUE "
610 PRINT M$;"JE REFLECHIS"
620 M$="484A4B484A4748454545"
630 FOR I=1 TO LEN(M$)-1 STEP 2
640 D=VAL(MID$(M$,I,1))
650 N=VAL("#"+MID$(M$,I+1,1))
660 MUSIC 1,D,N,0
670 PLAY 1,0,5,4000
680 WAIT 150
690 NEXT I
695 PLAY 0,0,0,0
700 RETURN
785 REM
790 REM * COMBINAISON *
795 REM
800 PLOT 1,15,10
810 PLOT 1,16,10
825 REM * * *
830 FOR R=1 TO 7
835 REM * *
840 FOR I=1 TO K
850 NZ=RND(1)
860 NEXT I
865 REM * *
870 A(R)=INT(NZ*49)+1
880 IF R=1 THEN 915
885 DJ=0
887 REM * *
890 FOR J=1 TO R-1
900 IF A(J)=A(R) THEN DJ=1
910 NEXT J
911 REM * *
912 IF DJ=1 THEN 840
915 IF R=7 THEN 925

```


```
920 PLOT 6*(R-1)+3,15,STR$(A(R))
921 PLOT 6*(R-1)+3,16,STR$(A(R))
922 IF R<7 THEN 930
925 M$="No COMPLEMENTAIRE "
928 PLOT 2,19,CHR$(10)+M$+STR$(A(R))
929 PLOT 2,20,CHR$(10)+M$+STR$(A(R))
930 NEXT R
932 REM * * *
935 POKE 616,24
940 PRINT
950 RETURN
```

12. Mnémonique téléphonique

Nombre de joueurs : de 1 à 78

Présentation :

Nos numéros de téléphone sont maintenant exclusivement numériques alors que quelques années auparavant, les parisiens pouvaient disposer de BALZAC 90 50 ou TRUDAINE 12 34. Pour les nostalgiques des lettres et les allergiques aux chiffres voici un programme qui vous propose des codages littéraux de votre numéro de téléphone.



Remarques :

Le tableau A\$ est rempli avec les 10 combinaisons de 3 lettres qui figurent aux lignes 50 et 60.

La ligne 140 extrait l'une quelconque des 3 lettres de la combinaison qui correspond à un chiffre du numéro et l'ajoute au début du codage déjà effectué.

Cette ligne 140 est exécutée autant de fois que le numéro à coder contient de caractères car elle figure au sein d'une boucle qui commence à la ligne 110 et se termine en 150.

La ligne 155 est chargée d'ajouter des espaces derrière le codage obtenu jusqu'à obtenir une longueur totale de 10 caractères (valeur imposée à LL en ligne 102) et cela pour des raisons de présentation sur l'écran.

Suggestion :

Supprimer des lignes de contrôle ou de formatage afin de comprendre leurs fonction : (ligne 103 ou 116 ou 155 ou 157 ou 170).

Programme :

```
1 REM
2 REM
3 REM
4 REM ** MNEMONIQUE **
5 REM ** TELEPHONIQUE **
6 REM -----
7 REM
8 REM
10 CLS
14 PAPER 6
16 INK 0
20 FOR I=0 TO 9
30 READ A$(I)
40 NEXT I
50 DATA DDQ,ZZZ,ABC,DEF,GHI
60 DATA JKL,MMN,PRS,TUV,WXY
70 PRINT:PRINT:PRINT
75 Z$=CHR$(138)+CHR$(147)+CHR$(132)
76 M$=" TAPEZ VOTRE NUMERO : "
80 PRINT Z$;M$;CHR$(150)
81 PRINT Z$;M$;CHR$(150)
83 PRINT:PRINT:PRINT
85 INPUT NB$
90 K=LEN(NB$)
95 PRINT
100 R$=""
102 LL=10
103 IF F/2=INT(F/2) THEN LL=9
104 REM
105 REM * MOT *
106 REM
110 FOR I=1 TO K
114 C$=MID$(NB$,I,1)
116 IF C$=" " THEN 150
120 C=VAL(C$)
130 H=INT(3*RND(1))+1
140 R$=R$+MID$(A$(C),H,1)
150 NEXT I
152 REM
```

```

153 REM * NORMALISE LA LONGUEUR A 10 *
154 REM
155 IF LEN(R$)<LL THEN R$=R$+" ":GOTO 155
157 IF F/4=INT(F/4) THEN PRINT
160 PRINT R$;
164 REM
165 REM * BAS DE L'ECRAN *
166 REM
170 IF PEEK(616)>21 THEN 180
175 F=F+1;GOTO 100
180 PLOT 1,23,CHR$(14)+CHR$(1)+CHR$(18)
182 PLOT 4,23,"en voulez-vous d'autres ? "
184 PLOT 1,24,CHR$(14)+CHR$(1)+CHR$(18)
186 PLOT 4,24,"en voulez-vous d'autres ? "
187 PLOT 30,23,22
188 PLOT 30,24,22
190 K$=KEY$
200 IF K$="" THEN 190
210 IF K$="N" THEN 300
220 F=0
230 POKE 616,11
240 FOR I=1 TO 15
250 PRINT CHR$(14)
260 NEXT I
270 POKE 616,10
280 GOTO 100
288 REM
289 REM * EFFACE 7 LIGNES *
290 REM
300 FOR I= 1 TO 6
310 PRINT CHR$(14)
320 NEXT I
330 END

```

13. Le mot mité

Nombre de personnes : 2

Présentation :

L'un des joueurs tape un mot à l'insu de l'autre car il ne s'inscrit pas sur l'écran.

L'ordinateur remplace les voyelles (A, E, I, O, U ou Y) par des étoiles.

Le mot ainsi "mité" est proposé à l'écran.

Le second joueur doit alors proposer des voyelles en les tapant au clavier (sans presser RETURN).

Les propositions successives remplaceront les étoiles dans l'ordre où elles apparaissent dans le mot.

Sitôt la dernière étoile remplacée, le verdict de l'ordinateur tombera.

Si vous ne vous en sortez vraiment pas, vous avez la ressource de presser la touche ?.



Remarques :

La sous-routine de détection des voyelles (500 à 560) est appelée 2 fois à des fins différentes :

— par la ligne 125 afin de remplacer une éventuelle voyelle par une étoile

— par la ligne 165 afin de vérifier que la touche pressée correspond bien à une voyelle.

La sous-routine n'exécute pas ces tâches mais lève (ou arbore) un

indicateur (encore appelé flag ou drapeau) afin qu'au retour, le programme appelant sache s'il y a voyelle ou pas.

Cet indicateur est la variable V qui est retournée par la sous-routine à 1 s'il y a voyelle ou à 0 dans le cas contraire.

Suggestion :

Modifier le programme afin qu'il masque non plus les voyelles mais les consonnes ou bien tout autre groupe de lettres défini à la ligne 560 !

Programme :

```
1 REM
2 REM
3 REM
4 REM *** LE MOT MITE ***
5 REM      -----
6 REM
8 REM
10 CLS
12 INK 3
15 PAPER 4
16 DIM A$(18)
17 PRINT
18 M$=" TAPEZ VOTRE MOT (18 car. maxi)"
20 H$=CHR$(138)
25 GOSUB 1000
27 PRINT:PRINT:PRINT
29 PRINT CHR$(19)
30 INPUT M0$
31 PRINT CHR$(19)
32 M=LEN(M0$)
34 IF M>18 THEN PING:CLEAR:GOTO 10
35 PLOT 1,1,10
37 PLOT 1,2,10
55 CLS
57 PRINT
58 M$="          DECHIFFREZ CE MOT"
60 GDSUB 1000
63 PRINT:PRINT
64 M$="DANS LEQUEL IL MANQUE LES VOYELLES"
```

```

65 GOSUB 1000
70 REM
80 REM * SUPPRESSION DES VOYELLES *
90 REM
100 PRINT:PRINT:PRINT
105 V$=""
106 W$=""
107 PRINT " ";
110 FOR Z=1 TO M
120 L$=MID$(M0$,Z,1)
125 GOSUB 500
130 IF V=1 THEN V$=V$+L$:L$="*"
135 A$(Z)=L$
140 PRINT L$+" ";
145 NEXT Z
150 PRINT CHR$(13);" ";
155 FOR Z=1 TO M
157 IF A$(Z)<>"*" THEN L$=A$(Z):GOTO 180
160 GET L$
162 IF L$="?" THEN 300
165 GOSUB 500
170 IF V=0 THEN PING:GOTO 160
175 W$=W$+L$
180 PRINT L$;" ";
210 NEXT Z
220 IF V$<>W$ THEN EXPLODE:GOTO 100
222 REM
223 REM * MUSIQUE *
224 REM
225 FOR I=1 TO 3
230 N$="313538413835"
235 FOR J=1 TO LEN(N$)-1 STEP 2
240 D=VAL(MID$(N$,J,1))
245 N=VAL("#"+MID$(N$,J+1,1))
250 MUSIC 1,D,N,0
255 PLAY 1,0,1,2000
260 WAIT 10
265 NEXT J
275 NEXT I
280 MUSIC 1,3,1,12
282 WAIT 50

```

```
285 PLAY 0,0,0,0
290 END
300 CLS
305 PRINT
355 PLOT 2,1,10
356 PLOT 2,2,10
360 PRINT:PRINT
365 M$=" LE MESSAGE ETAIT : "
370 GOSUB 1000
390 PRINT:PRINT
400 M$=M$+CHR$(150)+CHR$(129)
405 M$=M$+M0$+" "+CHR$(148)
410 GOSUB 1000
420 END
494 REM
495 REM * DETECTION DES VOYELLES *
496 REM
500 RESTORE
505 V=0
510 FOR I=1 TO 6
520 READ P$
530 IF L$=P$ THEN V=1
540 NEXT I
550 RETURN
560 DATA A,E,I,O,U,Y
994 REM
995 REM * AFFICHAGE *
996 REM
1000 PRINT H$;M$
1010 PRINT H$;M$
1020 RETURN
```


14. Que feriez-vous si...

Nombre de joueurs : 2 à 82

Présentation :

Chaque joueur à son tour pose une question et tape la réponse qu'il donnerait à sa question.

Après que tous les joueurs aient ainsi alimenté l'ordinateur, ce dernier restitue des couples question-réponse mais qui ne se correspondent plus ; cela donne souvent lieu à de très plaisantes rencontres.



Remarques :

La fin de question proposée par le joueur n'est pas acquise globalement à l'aide d'un INPUT mais caractère par caractère grâce au GET de la ligne 530.

Jusqu'à ce qu'il s'agisse du caractère correspondant à la touche RETURN les caractères ainsi saisis sont "concaténés" dans FQ\$ et le résultat est mémorisé dans une case du tableau B\$.

Le rôle de la ligne 750 est de choisir une réponse à associer à une question.

Suggestions :

Acquérir la fin de la question à l'aide de INPUT au lieu de GET.

Modifier la formule de choix de la ligne 750 afin que le décalage entre question et réponse soit de 2 au lieu de 1.

(Merci à Serge Pouts Lajus qui avait réalisé une version de ce programme sur un autre micro-ordinateur).

Programme :

```
1 REM
2 REM
3 REM
4 REM ** QUE FERIEZ-VOUS SI... **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
20 A$="QUE FERIEZ-VOUS SI "
25 Q$=" completez la question"
30 Q1$="dites ce que vous feriez si "
40 M$="NOMBRE DE JOUEURS"
50 PLOT 4,5,14:PLOT 8,5,M$
51 PLOT 4,6,14:PLOT 8,6,M$
55 POKE 616,8
56 PRINT
57 INPUT N
60 GOSUB 500:REM * CHOIX *
70 CLS
75 M$="* RESULTATS *"
80 PLOT 11,3,CHR$(10)+M$+CHR$(22)
81 PLOT 11,4,CHR$(10)+M$+CHR$(22)
85 M$="pour obtenir une question-reponse"
90 PLOT 1,7,CHR$(10)+M$+CHR$(22)
91 PLOT 1,8,CHR$(10)+M$+CHR$(22)
95 M$="il faut presser une touche.      "
100 PLOT 1,11,CHR$(10)+M$+CHR$(22)
101 PLOT 1,12,CHR$(10)+M$+CHR$(22)
120 GOSUB 700:REM * MELANGE *
130 END
485 REM
490 REM * CHOIX *
495 REM
```

```

500 FOR I=1 TO N
502 CLS
503 M$=CHR$(10)+"JOUER No"+CHR$(14)
504 PLOT 9,5,CHR$(1)+M$+MID$(STR$(I),2)
505 PLOT 9,6,CHR$(1)+M$+MID$(STR$(I),2)
510 PLOT 1,12,CHR$(4)+Q$
520 PLOT 1,15,CHR$(19)+A$+CHR$(22)
525 POKE 616,15
526 PRINT
528 POKE 617,25
530 GET L$
531 PRINT L$;
532 IF L$=CHR$(13) THEN 540
535 FQ$=FQ$+L$
538 GOTO 530
540 B$(I)=FQ$
545 POKE 616,19
547 PRINT
550 IF RIGHT$(FQ$,1)<>"?" THEN 555
552 FQ$=LEFT$(FQ$,LEN(FQ$)-1)
555 PRINT Q1$+FQ$
560 POKE 616,23
565 PRINT
580 INPUT C$(I)
585 FQ$=""
590 NEXT I
600 RETURN
685 REM
690 REM * MELANGE *
695 REM
700 FOR I=1 TO N
710 GET B$
720 CLS
730 PRINT:PRINT:PRINT
732 F$=" ?"
736 IF RIGHT$(B$(I),1)="?" THEN F$=""
740 PRINT A$;B$(I);F$
750 J=I-INT(I/N)*N+1
760 GET B$
765 PRINT:PRINT:PRINT
770 L=PEEK(616)

```

```
775 PRINT C$(J)  
780 NEXT I  
790 RETURN
```

15. Ration calorique

Nombre de personnes : 1

Présentation :

Votre forme dépend pour une part de la façon dont vous vous nourrissez et, dans ce domaine, du respect de la ration calorique qui vous est nécessaire : ni trop, ni trop peu, voilà la règle ! Mais... comment savoir ?

Le programme qui suit détermine pour vous le nombre de calories journalières qui convient à une personne ayant votre profil ; laissez-vous guider par les questions et ... répondez sincèrement. Attention ! si l'ordinateur vous accorde 2 300 calories, ne les absorbez tout de même pas sous la forme de 300 g de mayonnaise ou de 600 g de sucre !

Un dernier point, le programme n'acceptera de considérer votre cas qui si vous avez 10 ans au moins.

CALCUL DE VOTRE RATION CALORIQUE

VOTRE AGE (en ans)	? 37
VOTRE POIDS (en kg)	? 72
VOTRE TAILLE (en cm)	? 178
VOTRE SEXE (M ou F)	? M
VOTRE ACTIVITE (0 à 2000)	? 12

VOS BESOINS ENERGETIQUES :
2785 CALORIES PAR JOUR.

Ready

Remarques :

La formule qui permet de déterminer les besoins liés à votre métabolisme de base tient compte de la surface de peau que vous ne connaissez probablement pas, mais dont l'ordinateur calcule une valeur approchée qui est fonction de votre taille et de votre poids (la formule figure en ligne 75).

A la fin du programme, vous pourrez connaître la surface de votre peau en tapant au clavier :

PRINT S RETURN

Dans le chapitre ACTIVITÉ, vous n'êtes pas tenus aux multiples

de 5 qui apparaissent en exemple et vous pouvez fort bien vous évaluer à 12.

L'ensemble des âges est découpé en 5 tronçons :

- 0 à 9 ans
- 10 à 15 ans
- 16 à 20 ans
- 21 à 60 ans
- au-delà de 60 ans

Les lignes 39 à 46 ont pour rôle de détecter celle de ces catégories qui vous concerne et de prendre une décision en conséquence.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** RATION CALORIQUE **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
20 X=0:Y=1:AT=10:FD=21:AV=0:FL=22
21 M$="CALCUL DE VOTRE RATION CALORIQUE"
22 NB=2
23 GOSUB 1000
25 REM * * * * *
30 Y=5:FD=19:AV=4
31 M$="VOTRE AGE (en ans) "
32 GOSUB 1000
35 C=29:L=6
36 GOSUB 2000
37 AG=VAL(R$)
39 IF AG<10 THEN ZAP:GOTO 35
40 IF AG>9 AND AG<16 THEN A=45
42 IF AG>15 AND AG<21 THEN A=40
44 IF AG>20 AND AG<61 THEN A=35
46 IF AG>60 THEN A=30
47 REM * * * * *
```

```

50 Y=7
51 M$="VOTRE POIDS      (en kg ) "
52 GOSUB 1000
55 L=8
56 GOSUB 2000
57 M=VAL(R$)
58 REM * * * * *
60 Y=9
61 M$="VOTRE TAILLE    (en cm ) "
62 GOSUB 1000
65 L=10
66 GOSUB 2000
67 TC=VAL(R$)
68 IF TC<3 THEN ZAP:GOTO 65
69 T=TC/100
70 REM
71 REM * SURFACE DE VOTRE PEAU *
72 REM
75 S=0.2*M^0.4*T^0.7
77 REM
78 REM * CALCUL DU METABOLISME DE BASE *
79 REM
80 MB=(24*A-1)*S
85 REM * * * * *
90 Y=11
91 M$="VOTRE SEXE      (M ou F) "
92 GOSUB 1000
95 L=12
96 GOSUB 2000
97 S$=R$
100 IF S$="M" THEN 200
110 IF S$<>"F" THEN ZAP:GOTO 95
115 REM * * * * *
120 Y=13
121 M$="ENCEINTE ?    (O ou N) "
122 GOSUB 1000
125 L=14
126 GOSUB 2000
127 EN$=R$
130 IF EN$="O" THEN MB=MB+600:GOTO 200
140 IF EN$<>"N" THEN ZAP:GOTO 125

```

```

150 MB=(MB*4)/5
190 REM * * * * *
200 Y=15
201 M$="VOTRE ACTIVITE (0 a 20) "
202 GOSUB 1000
210 X=3:Y=17:FD=23:AV=1
211 M$="TOTALEMENT INACTIF --> 0 "
212 GOSUB 1000
220 Y=19
221 M$="ACTIVITE REDUITE --> 5 "
222 GOSUB 1000
230 Y=21
231 M$="ACTIVITE LEGERE --> 10 "
232 GOSUB 1000
240 Y=23
241 M$="ACTIVITE CERTAINE --> 15 "
242 GOSUB 1000
250 Y=25
251 M$="FORTE ACTIVITE --> 20 "
252 GOSUB 1000
260 L=16
265 GOSUB 2000
266 REM
267 REM * CALCUL DU METABOLISME D'ACTIVITE *
268 REM
270 MA=VAL(R$)
275 IF MA<0 OR MA>20 THEN ZAP:GOTO 260
276 MA=80*MA+400
277 REM
278 REM * EFFACEMENT PARTIEL DE L'ECRAN *
279 REM
280 REM
284 POKE 616,18
285 FOR I=1 TO 9
290 PRINT CHR$(14)
300 NEXT I
310 PRINT CHR$(14);
314 REM
315 REM * METABOLISME TOTAL *
316 REM
320 MT=INT(MA+MB)

```

```

330 X=2:Y=19
331 M$="VOS BESOINS ENERGETIQUES : "
332 GOSUB 1000
340 Y=21
341 M$=" CALORIES PAR JOUR. "
342 M$=MID$(STR$(MT),2)+M$
343 GOSUB 1000
970 POKE 616,23:PRINT
980 END
989 REM
990 REM ** AFFICHAGE **
991 REM
1000 N$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1010 FOR NN=1 TO NB
1020 PLOT X,Y+NN-1,N$+M$+CHR$(FL)
1030 NEXT NN
1040 RETURN
1989 REM
1990 REM ** SAISIE D'UNE INFORMATION **
1991 REM
2000 PLOT C,L,8
2010 POKE 616,L
2020 PRINT
2030 POKE 617,C+2
2040 INPUT R$
2050 RETURN

```

16. La rencontre

Nombre de personnes : 2 à 79

Présentation :

L'ordinateur vous propose le squelette d'une courte histoire : celle d'une rencontre.

Chaque joueur tour à tour donne à l'ordinateur des précisions concernant les personnages, les circonstances, les comportements.

L'histoire aura toujours la structure suivante :

Untel RENCONTRA Unetelle à tel endroit pour telle raison (ou dans tel but).

Blablabla DIT Untel

Blablablabla REPONDIT Unetelle


Il fit ceci

Elle fit cela

Il se produisit telle chose

(Merci à Serge Pouts Lajus pour le scénario de cette histoire).

Pour peu que Untel et Unetelle désignent des personnes parmi les joueurs, le rire est assuré.



Remarques :

Il faudra, bien entendu, que chaque joueur entre en mémoire ce qu'il aura écrit en pressant la touche RETURN.

L'essentiel du programme tient dans 2 tableaux :

A\$ contient les 9 messages qui invitent les joueurs à répondre au clavier.

B\$ contient les 9 propositions tapées par les joueurs (lignes 120 à 180).

Proposer le récit est alors enfantin puisqu'il suffit de "vider" le tableau B\$ sur l'écran en prenant cependant quelques précautions pour que la forme et la présentation des phrases soient convenables (lignes 230 à 310).

Programme :

```
1 REM
2 REM
3 REM
4 REM ** LA RENCONTRE **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
11 CLEAR
12 PAPER 6
15 INK 1
17 X=12:Y=5:AT=14:FD=19:AV=1:FL=22
18 NB=2:M$="ATTENTION "
19 GOSUB 1000
20 X=2:Y=9:AT=10
22 M$="UN SEUL JOUEUR REGARDE L'ECRAN "
23 GOSUB 1000
25 REM
26 REM * MESSAGES *
27 REM
30 A$(1)="Tapez le nom d'une femme "
35 A$(2)="Tapez le nom d'un homme "
37 M$="Ou la rencontre "
40 A$(3)=M$+"a-t-elle lieu ? "
45 M$="Pourquoi "
50 A$(4)=M$+"se sont-ils rencontrés ? "
60 A$(5)="Qu'a dit l'homme ? "
70 A$(6)="Qu'a répondu la femme ? "
80 A$(7)="Qu'a fait l'homme ? "
90 A$(8)="Qu'a fait la femme ? "
100 A$(9)="Qu'advint-il ensuite ? "
```

```

110 X=2:Y=19:AT=8:FD=22:AV=4:FL=22
111 NB=1:M$=" Appuyez sur une touche "
112 GOSUB 1000
115 GET Q$
117 REM
118 REM * REPONSES DES JOUEURS *
119 REM
120 FOR I=1 TO 9
125 CLS
130 X=2:Y=9:AT=10:FD=22:AV=4:FL=22
131 NB=2:M$=A$(I)
132 GOSUB 1000
140 Y=19
141 NB=2
142 M$=" puis laissez "
143 M$=M$+" le clavier au suivant"
145 GOSUB 1000
150 POKE 616,15
155 PRINT
160 INPUT B$(I)
180 NEXT I
190 CLS
195 REM
197 REM * LE RECIT *
199 REM
200 X=2:Y=9:AT=10:FD=21:AV=3:FL=22
201 NB=2:M$=" VOUS OBTIENDREZ LE RECIT "
202 GOSUB 1000
205 Y=11
206 M$=" EN PRESSANT UNE TOUCHE "
207 GOSUB 1000
210 GET Q$
220 REM
230 CLS
240 POKE 616,5
245 PRINT
250 PRINT B$(1);" rencontra ";B$(2);
252 PRINT " ";B$(3);" ";B$(4)
255 PRINT
260 PRINT B$(5);" dit ";B$(2)
265 PRINT

```

```
270 PRINT B$(6);" repondit " ;B$(1)
275 PRINT
280 PRINT B$(7)
285 PRINT
290 PRINT B$(8)
295 PRINT
300 PRINT B$(9)
310 END
985 REM
990 REM * AFFICHAGE *
995 REM
1000 D$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1005 FOR NN=1 TO NB
1010 PLOT X,Y+NN-1,D$+M$+CHR$(FL)
1020 NEXT NN
1030 RETURN
```

17. Rotation

Pour 1 joueur
(ou 2 si on invente une nouvelle règle)

Présentation :

Les 25 premières lettres de l'alphabet vous sont proposées en désordre dans un carré de 5 sur 5.

Il s'agit pour vous de les remettre dans l'ordre alphabétique par ligne :

A B C D E
F G H I J
K L M N O
P Q R S T
U V W X Y


La seule possibilité de modification dont vous disposez est une rotation de un quart de tour vers la droite des 4 lettres d'un carré de 2 lettres sur 2 lettres. Le coin supérieur gauche de ce carré est un caractère apparaissant en affichage "inversé". (Il apparaît par exemple écrit bleu clair sur bleu foncé alors que les autres sont bleu foncé sur bleu clair). Cette rotation s'obtient en pressant la barre d'espace.

Exemple :

AVANT
D U G K O
S A J C V
E W P N M
T L F B I
R H X Q Y

APRÈS
D U G K O
S W A C V
E P J N M
T L F B I
R H X Q Y

Enfin, vous pourrez déplacer la position de la lettre "inversée" à l'aide des 4 touches portant des flèches. Notez cependant que vous ne pourrez toucher ainsi que 16 caractères sur les 25 puisque cette lettre représente le coin supérieur gauche d'un carré de 2 sur 2.



Remarques :

Les tableaux T\$ et U\$ déclarés aux lignes 35 et 36 ne servent qu'au mélange des 25 lettres.

Au cours du jeu proprement dit, la reconnaissance des caractères se fait par test sur l'écran (instruction SCRN).

Suggestion :

Au lieu de la rotation de 4 caractères proposée, on peut envisager une rotation des 8 caractères qui entourent le caractère inversé.

Dans ce cas il faudra limiter la position de l'inversion aux 9 caractères du carré central et adapter la routine de rotation (lignes 500 à 800).

Programme :

```

1 REM
2 REM
3 REM
4 REM ** ROTATION **
5 REM -----
6 REM
7 REM
8 REM
10 CLS

```

```

12 PAPER 6
15 INK 4
17 M$=CHR$(10)+CHR$(19)+CHR$(1)
18 PLOT 8,1,M$+"R O T A T I O N "+CHR$(22)
19 PLOT 8,2,M$+"R O T A T I O N "+CHR$(22)
20 POKE 616,15
21 PRINT CHR$(17)
22 M$="poussez une touche "+CHR$(22)
25 PRINT CHR$(140)+CHR$(128)+M$
27 A=RND(1)
28 IF KEY$="" THEN 27
30 PRINT CHR$(11);CHR$(14)
35 DIM T$(24)
36 DIM U$(4,4)
40 FOR I=0 TO 24
45 T$(I)=CHR$(65+I)
50 NEXT I
60 FOR I=1 TO 20
70 R=INT(RND(1)*24+1)
80 S=INT(RND(1)*24+1)
90 V$=T$(R)
100 T$(R)=T$(S)
110 T$(S)=V$
115 PLOT 1,I+2,10
120 NEXT I
130 FOR L=0 TO 4
140 FOR C=0 TO 4
145 TC$=T$(5*L+C)
150 PLOT 10+C*4,5+L*4,TC$
155 PLOT 10+C*4,6+L*4,TC$
160 U$(C,L)=TC$
165 NEXT C
170 NEXT L
175 POKE 616,23
176 PRINT
180 C=0
185 L=0
190 SC=SCRN(10+C*4,5+L*4)+128
194 PLOT 10+C*4,5+L*4,CHR$(SC)
198 PLOT 10+C*4,6+L*4,CHR$(SC)
200 GET A$

```

```

205 IF A$="F" THEN PRINT CHR$(17):END
210 A=ASC(A$)-7
220 IF A=25 THEN 500
224 CP=C
228 LP=L
230 IF A<1 OR A>4 THEN ZAP:GOTO 200
240 ON A GOTO 250,260,270,280
250 IF C=0 THEN ZAP:GOTO 200
252 C=C-1
255 GOTO 300
260 IF C=3 THEN ZAP:GOTO 200
262 C=C+1
265 GOTO 300
270 IF L=3 THEN ZAP:GOTO 200
272 L=L+1
275 GOTO 300
280 IF L=0 THEN ZAP:GOTO 200
282 L=L-1
300 SC=SCRN(10+CP*4,5+LP*4)-128
310 PLOT 10+CP*4,5+LP*4,CHR$(SC)
320 PLOT 10+CP*4,6+LP*4,CHR$(SC)
330 GOTO 190
494 REM
495 REM * ROTATION *
496 REM
500 CC=SCRN(10+C*4,5+L*4)-128
505 RESTORE
510 FOR I=1 TO 3
520 READ CS,LS,CB,LB
530 SC=SCRN(10+(C+CS)*4,5+(L+LS)*4)
540 PLOT 10+(C+CB)*4,5+(L+LB)*4,CHR$(SC)
550 PLOT 10+(C+CB)*4,6+(L+LB)*4,CHR$(SC)
555 NEXT I
560 PLOT 10+(C+1)*4,5+(L*4),CHR$(CC)
570 PLOT 10+(C+1)*4,6+(L*4),CHR$(CC)
575 GOTO 190
580 SC=SCRN(10+C*4,5+L*4)+128
590 PLOT 10+C*4,5+L*4,CHR$(SC)
600 PLOT 10+C*4,6+L*4,CHR$(SC)
610 GOTO 190
800 DATA 0,1,0,0,1,1,0,1,1,0,1,1

```

18. Solitaire

Nombre de joueurs : 1 bien entendu !

Présentation :

Il s'agit du jeu de solitaire bien connu. Quelques explications cependant pour ceux qui ne le connaîtraient pas.

Sur un terrain en forme de croix il y a 33 emplacements dont 32 sont occupés par des pions (○), l'emplacement central étant libre (.).


La seule chose que puisse faire un pion, c'est sauter par-dessus un pion voisin (sur la même ligne ou dans la même colonne) afin de retomber dans un emplacement libre.



Le but du jeu : éliminer 31 des 32 pions.

Le programme vous demande tour à tour les coordonnées du pion sauteur puis celles de la case réceptrice.

La validité du coup est contrôlée par l'ordinateur qui rejette toute tentative illégale et qui exécute toute demande correcte.



Remarques :

Même si le sujet s'y prête parfaitement, le programme n'utilise pas les tableaux et effectue tous ses contrôles de légalité à partir de tests qu'il fait sur l'écran (instruction SCRN).

Ce qui concerne le tableau T (lignes 45, 90 et 120) peut être purement et simplement supprimé et n'est là qu'à titre indicatif pour ceux qui voudraient réaliser un programme similaire avec tests sur les contenus d'un tableau qui serait le reflet de la situation sur le terrain de jeu.

Pour saisir le pourquoi des lignes 214 et 216, vous n'avez qu'à les supprimer et observer les différences au moment de l'exécution.

Suggestions :

Modifier le programme afin que le trou de départ ne soit pas au centre mais une case au choix du joueur.

Mémoriser en DATA (à partir de la ligne 4000) une partie modale qui sera exécutée en guise de démonstration par une routine à écrire vous-même.

Proposer un terrain de jeu prenant moins de place sur l'écran (étudier le rôle du facteur 3 dans les lignes 50, 60, 140, 170, 175, 1100 et 1110).

Programme :

```
1 REM
2 REM
3 REM
4 REM ** SOLITAIRE **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 0
15 INK 6
16 REM
17 REM ** DESSIN DES PIONS **
18 REM
20 FOR L=0 TO 6
30 FOR C=0 TO 6
40 IF L>1 AND L<5 OR C>1 AND C<5 THEN 50
45 T(C,L)=9:GOTO 100
50 X=3*C+9
60 Y=3*L
70 PLOT X,Y,"O"
```

```

80 IF C=L=9 THEN PLOT X,Y,"."
90 T(C,L)=1
100 NEXT C
110 NEXT L
114 REM
115 REM ** ECRITURE DES COORDONNEES **
116 REM
120 T(3,3)=0
130 FOR I=65 TO 71
140 PLOT 9+(I-65)*3,20,CHR$(I)
150 NEXT I
160 FOR I=49 TO 55
170 PLOT 5,(55-I)*3,CHR$(I)
175 PLOT 6,(55-I)*3,3
180 NEXT I
189 REM
190 REM ** PARTIE **
191 REM
200 M$="PIDN"
210 GOSUB 1000
211 IF ER=1 THEN 2000
212 IF SCRN(C,L) <> 79 THEN 2000
214 PLOT C+1,L,8
216 PLOT C-1,L,12
220 CP=C
230 LP=L
235 F=1
240 M$="TROU"
250 GOSUB 1000
252 IF ER=1 THEN 2000
255 IF SCRN(C,L) <> 46 THEN 2000
260 CT=C
270 LT=L
280 IF CP <> CT AND LP <> LT THEN 2000
290 DC=ABS(CP-CT)
300 DL=ABS(LP-LT)
310 IF DL*DC <> 0 OR DL+DC <> 6 THEN 2000
320 CI=(CP+CT)/2
330 LI=(LP+LT)/2
340 IF SCRN(CI,LI) <> 79 THEN 2000
350 PLOT CI,LI,"."

```

```

355 PLOT CP-1,LP,8
360 PLOT CP,LP,"."
370 PLOT CT,LT,"0"
380 PD=PD+1
390 IF PD<32 THEN 200
890 GET A$
900 END
989 REM
990 REM ** ACQUISITION DES COORDONNEES **
991 REM
1000 POKE 616,22
1002 PRINT
1003 N$=CHR$(8)
1004 PLOT 3,25,CHR$(10)+CHR$(2)+M$
1005 PLOT 3,26,CHR$(10)+CHR$(2)+M$
1007 PLOT 13,25,N$+"colonne (A a G)      "
1010 PLOT 13,26,N$+"ligne   (1 a 7)      "
1015 POKE 616,25
1018 PRINT
1020 POKE 617,32
1030 GET C$
1035 IF C$="0" THEN 3000
1040 PRINT C$
1050 POKE 616,26
1060 PRINT
1070 POKE 617,32
1080 GET L$
1085 IF C$="0" THEN 3000
1090 PRINT L$;
1100 C=(ASC(C$)-65)*3+9
1110 L=(55-ASC(L$))*3
1115 ER=0
1120 IF C<9 OR C>27 THEN ER=1
1130 IF L<0 OR L>18 THEN ER=1
1900 RETURN
1990 REM
1991 REM ** ERREUR **
1992 REM
2000 IF F=0 THEN CP=C:LP=L
2010 ZAP
2015 PLOT CP-1,LP,8

```

```
2020 GOTO 200  
3000 POKE 616.0  
3010 PRINT  
3020 END
```

19. Synthétiseur

Pour 1 personne

Présentation :

Ce programme transforme votre ORIC en un instrument de musique vous permettant de jouer n'importe quelle note sur 6 octaves.

Les touches "blanches" sont

R T Y U I O P

qui génèrent respectivement les notes

DO RÉ MI FA SOL LA SI

Quant aux touches **5 6 8 9 10** elles donnent accès aux dièses et aux bémols.

Les touches **Q** et **W** permettent le changement d'octave

Q = plus grave

W = plus aigu

Les touches **A** et **S** permettent de modifier le volume

A = plus faible

S = plus fort

ESC (et bien d'autres touches !) arrête l'émission de son.

Enfin, vous presserez la touche marquée **/** pour quitter ce programme.



Merci à Damien Thomassin de *Quintefeuille informatique* qui est à l'origine de ce programme.

Suggestion :

L'émission des notes est obtenue à la ligne 660 à l'aide de la seule instruction MUSIC.

Il est possible, en fixant le volume à 0

660 MUSIC 1,OCT,NT,0

de transférer le contrôle à l'instruction PLAY pour profiter de la diversité qu'elle offre sur "l'enveloppe" du son produit.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** SYNTHETISEUR **
5 REM      -----
6 REM
7 REM * auteur : Damien THOMASSIN *
8 REM
10 CLS
12 PAPER 0
15 INK 6
20 REM
25 REM * REDEFINITION DU TIRET *
27 REM
30 A=ASC("-")*8+46080+3
40 POKE A,63
45 PRINT
50 REM
55 REM * PRESENTATION *
60 REM
61 K$="-----"
62 OM$=K$+"      ORIC MUSICAL      "+K$
63 FOR I=1 TO 37
65 L$=L$+"-"
67 NEXT I
70 PRINT CHR$(131);L$
90 PRINT CHR$(138);CHR$(131);OM$
100 PRINT CHR$(138);CHR$(131);OM$
120 PRINT CHR$(131);L$
122 REM
123 REM * FOND ROUGE SUR LES LIGNES 1 ET 2 *
```

```

124 REM
126 FOR I=0 TO 2
127 POKE 48123+40*I,17
128 NEXT I
130 REM
140 REM * INSTRUCTIONS *
150 REM
300 REM
310 REM * DESSIN DU CLAVIER *
315 REM
320 PRINT "          -   -   -   -   -"
325 PRINT "          \5\ \6\   \8\ \9\ \0\"
327 PRINT "          -----"
330 PRINT "          \R\ \T\ \Y\U\ \I\ \O\ \P\"
335 PRINT "          -   -   -   -   -"
340 PRINT "          ^   ^   ^ ^   ^   ^   ^"
350 PRINT
360 PRINT "          DO RE MI FA SOL LA SI"
363 FOR I=0 TO 7
365 POKE 48323+40*I,19
366 POKE 48324+40*I,4
367 NEXT I
370 PRINT
380 PRINT " 'W' -> octave superieure"
390 PRINT " 'Q' -> octave inferieure"
395 PRINT
400 PRINT " 'A' -> plus doux"
410 PRINT " 'S' -> plus fort"
415 PRINT
420 PRINT " 'ESC' -> silence"
440 PRINT " '/' -> sortie du programme"
450 PLOT 20,25,"Octave No   "+CHR$(16)
454 POKE 49053,18
456 POKE 49054,1
465 OCT=4
467 VOL=8
470 GET A$
475 REM
480 REM * TEST DU CARACTERE ENTRE *
485 REM
490 IF A$="R" THEN NT=1:GOTO 625

```

```

495 IF A$="5" THEN NT=2:GOTO 625
500 IF A$="7" THEN NT=3:GOTO 625
505 IF A$="6" THEN NT=4:GOTO 625
510 IF A$="Y" THEN NT=5:GOTO 625
515 IF A$="U" THEN NT=6:GOTO 625
520 IF A$="8" THEN NT=7:GOTO 625
525 IF A$="I" THEN NT=8:GOTO 625
530 IF A$="9" THEN NT=9:GOTO 625
535 IF A$="0" THEN NT=10:GOTO 625
540 IF A$="0" THEN NT=11:GOTO 625
545 IF A$="P" THEN NT=12:GOTO 625
560 IF A$="/" THEN END
570 IF A$="O" AND OCT>1 THEN OCT=OCT-1
580 IF A$="W" AND OCT<6 THEN OCT=OCT+1
590 IF A$="A" AND VOL>1 THEN VOL=VOL-1
595 IF A$="S" AND VOL<15 THEN VOL=VOL+1
600 GOTO 470
615 REM * AFFICHE LA NOTE ET L'OCTAVE *
625 RESTORE
630 FOR I=1 TO NT
632 READ W$
635 NEXT I
640 PLOT 14,25,W$+" "
645 PLOT 30,25,CHR$(OCT+48)+" "+CHR$(16)
650 REM
652 REM * EXECUTION DE LA NOTE *
655 REM
660 MUSIC 1,OCT,NT,VOL
680 GOTO 470
1985 REM
1990 REM * LISTE DES NOMS DES NOTES *
1995 REM
2000 DATA DO,DD#,RE,RE#,MI,FA
2010 DATA FA#,SOL,SOL#,LA,LA#,SI

```

20. Taquin

Nombre de joueurs : 1

Présentation :


Jeu de lettres très répandu sous forme d'un cadre en matière plastique à l'intérieur duquel on déplace des lettres dans le but de les remettre dans l'ordre alphabétique.

Ici, pas de problèmes matériels, les lettres sont sur l'écran.

Dans la version proposée vous disposez de 24 lettres et d'une case libre.

La seule chose qu'un joueur puisse faire, c'est de glisser dans le trou l'une des lettres qui le jouxtent, pour cela il suffit de presser la touche correspondant à la lettre à déplacer.

Pour sortir du jeu, il faut presser Y.



Remarques :

Dans ce jeu, j'ai mémorisé l'état du terrain de jeu dans un tableau linéaire T\$ de 25 cases (numérotées de 0 à 24). Un tableau à 2 dimensions (5 sur 5) aurait, bien entendu, pu être utilisé (voir suggestions).

Comme dans DACTYLORIC et ROTATION, les lettres sont d'abord écrites dans un tableau en ordre alphabétique (lignes 40 à 50) puis mélangées (lignes 60 à 120).

Suggestions :

Réaliser le même jeu avec 23 lettres + 1 trou disposés dans un rectangle de 6 sur 4.

Utiliser un tableau à 2 dimensions pour gérer le déroulement du jeu tel qu'il est proposé ; il faut commencer par remplacer la

ligne 35 par 35 DIMT\$ (4,4) et modifier pas mal de choses à la suite.

La fin du jeu est souvent rendue difficile à cause de 2 lettres qui sont côte à côte sur une même ligne, mais pas dans le bon ordre ! Pour remédier à cette situation agaçante, prévoir une possibilité de permutation des contenus de 2 cases contiguës, possibilité qui ne pourra être utilisée *qu'une fois au plus* au cours d'une partie !

Programme :

```
1 REM
2 REM
3 REM
4 REM ** TAQUIN **
5 REM    ---==
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
17 M$=CHR$(10)+CHR$(19)+CHR$(1)
18 PLOT 10,1,M$+"T A Q U I N "+CHR$(22)
19 PLOT 10,2,M$+"T A Q U I N "+CHR$(22)
20 POKE 616,15
21 PRINT CHR$(17)
22 M$="pressez une touche "+CHR$(22)
25 PRINT CHR$(140)+CHR$(128)+M$
27 A=RND(1)
28 IF KEY$="" THEN 27
30 PRINT CHR$(11);CHR$(14)
31 REM
32 REM * METTRE 25 LETTRES EN DESORDRE *
33 REM
35 DIM T$(24)
40 FOR I=0 TO 23
45 T$(I)=CHR$(65+I)
50 NEXT I
55 T$(24)="."
60 FOR I=1 TO 20
```

```

70 R=INT(RND(1)*24+1)
80 S=INT(RND(1)*24+1)
90 V$=T$(R)
100 T$(R)=T$(S)
110 T$(S)=V$
115 PLOT 1,I+2,10
120 NEXT I
124 REM
125 REM * AFFICHAGE DE LA SITUATION *
126 REM
130 FOR L=0 TO 4
140 FOR C=0 TO 4
150 PLOT 10+C*4,5+L*4,T$(5*L+C)
155 PLOT 10+C*4,6+L*4,T$(5*L+C)
160 NEXT C
170 NEXT L
179 REM
180 REM * CORPS DU JEU *
181 REM
200 GET A$
205 IF A$="Y" THEN END
210 IF A$<"A" OR A$>"Y" THEN ZAP:GOTO 200
220 FOR I=0 TO 24
230 IF T$(I)=A$ THEN CT=I
240 IF T$(I)="." THEN CP=I
250 NEXT I
260 DT=ABS(CP-CT)
270 IF DT<>1 AND DT<>5 THEN ZAP:GOTO 200
280 L1=INT(CT/5)
290 C1=CT-L1*5
300 L2=INT(CP/5)
310 C2=CP-L2*5
320 IF C1<>C2 AND L1<>L2 THEN ZAP:GOTO 200
330 T$(CT)="."
335 IF C1<>C2 AND L1<>L2 THEN ZAP:GOTO 200
340 T$(CP)=A$
350 PLOT 10+C1*4,5+L1*4,"."
360 PLOT 10+C1*4,6+L1*4,"."
370 PLOT 10+C2*4,5+L2*4,A$
380 PLOT 10+C2*4,6+L2*4,A$
390 GOTO 200

```


Programme :

```
1 REM
2 REM
3 REM
4 REM ** TELEPATHIE **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 0
15 INK 6
17 NN=20
18 PRINT CHR$(17)
20 DIM A(40)
25 R=0
27 MX=99
30 REM
40 REM * CHOIX DES NOMBRES *
50 REM
55 M$= CHR$(138)+CHR$(129)+CHR$(146)
60 PRINT M$;" EMETTEUR ";CHR$(144)
65 PRINT M$;" EMETTEUR ";CHR$(144)
70 PRINT:PRINT:PRINT
80 PRINT "ECRIVEZ UNE LISTE DE ";
81 PRINT NN;"NOMBRES"
82 PRINT
83 PRINT"COMPRIS ENTRE 10 ET 99"
84 PRINT
85 PRINT "SEPARES PAR DES";
86 PRINT CHR$(140);"'RETURN'."
87 PRINT:PRINT:PRINT
90 REM * ACQUISITION DE LA SEQUENCE *
91 REM
92 FOR I=1 TO NN
95 M$="Nombre No "
100 PRINT CHR$(131);M$;I;"      ";
105 POKE 617,16
110 INPUT A$
112 A=VAL(A$)
```

```

115 IF A=INT(A) AND A<>0 THEN 120
117 PRINT CHR$(13);CHR$(11);
118 ZAP
119 GOTO 100
120 IF A<10 OR A>99 THEN 117
125 A(I)=A
127 PRINT CHR$(11);
128 PING
130 NEXT I
140 CLS
150 REM
160 REM * CHOIX DU RECEPTEUR *
170 REM
175 PRINT
177 M$= CHR$(138)+CHR$(145)+CHR$(130)
180 PRINT M$;" RECEPTEUR ";CHR$(144)
185 PRINT M$;" RECEPTEUR ";CHR$(144)
186 PRINT:PRINT
187 PRINT:PRINT
188 M$="L'emetteur telepathique a choisi"
190 PRINT M$
192 PRINT
193 M$="nombres compris entre 10 et 99"
195 PRINT NN;M$
196 PRINT:PRINT:PRINT
197 M$=" A quels nombres a-t-il pense ?"
200 PRINT M$
205 PRINT:PRINT
210 FOR I=1 TO NN
215 M$="Nombre No "
220 PRINT CHR$(131);M$;I;" ";
225 POKE 617,16
230 INPUT A$
235 A=VAL(A$)
240 IF A=INT(A) AND A<>0 THEN 255
250 PRINT CHR$(13);CHR$(11);
252 ZAP
253 GOTO 220
255 IF A<10 OR A>99 THEN 250
257 POKE 617,22
260 PRINT CHR$(11);

```

```

265 M#=CHR$(140)+"Faux "
270 IF A<>A(I) THEN 280
275 M#=CHR$(136)+"Juste":R=R+1
280 PRINT "    ":A(I);M#
285 REM
290 REM * TEST DES 10 *
295 REM
300 V=I-INT(I/10)*10
302 IF V=0 THEN GOSUB 1000 ELSE 330
305 IF I=MN THEN PRINT CHR$(17):END
307 PRINT
310 PRINT " Voulez-vous continuer ? ";
315 GET R#
320 IF R#<>"D" THEN PRINT CHR$(17):END
325 PRINT
330 NEXT I
900 REM
910 REM * DONS TELEPATHIQUES *
920 REM
930 REM * CALCUL DE L'ECART-TYPE *
940 REM
1000 SIG=(1/SQR(I))*(SQR(MX-1)/MX)
1010 X=R/I
1020 F=1/MX+SIG
1060 REM
1070 REM * DISTANCE A L'ECART-TYPE *
1080 REM
1085 PRINT:PRINT
1090 IF ABS(X-1/MX)>=SIG THEN 1100
1092 M#="Pas de dons particuliers !"
1095 PRINT M#
1097 GOTO 1180
1100 IF X-1/MX>=-SIG THEN 1135
1102 M#="Votre malchance est etonnante !"
1105 PRINT M#
1107 GOTO 1180
1110 REM
1120 REM * NOTE DE TELEPATHE *
1130 REM
1135 PRINT:PRINT
1140 Y=50+50*(X-F)/(1-F)

```

```
1150 PRINT "Incontestablement vous avez"  
1155 PRINT "des dons de telepathe !"  
1157 PRINT  
1160 PRINT "Dans l'echelle des dons " ;  
1165 PRINT "telepathiques"  
1170 PRINT "qui s'etend de 50 a 100"  
1173 PRINT "votre niveau est " ;  
1175 PRINT "actuellement de " ; INT(Y)  
1180 RETURN
```

22. Test de Ruffier

Pour 1 personne à la fois


Présentation :

Votre faculté de “récupération cardiaque” est un élément qui permet de déterminer votre état de santé.

Le test comporte 3 prises de pouls :

- 1) au repos
- 2) immédiatement après un effort
- 3) 1 minute après l'effort

Des 3 mesures faites, l'ordinateur déduira votre forme physique.



Remarques :

La présentation de chacune des 3 phases est identique et les attributs de caractères correspondants figurent en ligne 21.

Dans la récapitulation des mesures, les mots et les nombres ne réapparaissent pas de la même couleur bien qu'on n'ait pas changé d'attribut entre les mots et les nombres.

En fait l'attribut couleur a été modifié par le nombre lui-même car un nombre positif est toujours précédé d'un caractère qui s'affiche en général sous la forme d'un espace.

Suggestion :

Si vous désirez éliminer le caractère “parasite” évoqué plus haut, remplacez le STR\$(P.) par des MID\$(STR\$(P.),2) dans les lignes 161, 171 et 181.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** TEST DE RUFFIER **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
16 REM
17 REM ** LE TEST **
18 REM
19 REM * 1ere phase *
20 REM
21 X=0:Y=3:AT=10:FD=19:AV=4:FL=22
22 NB=2
23 M$="PRENEZ VOTRE POULS AU REPOS "
24 GOSUB 1000
25 Y=5
26 M$="PUIS TAPEZ VOTRE NOMBRE DE "
27 GOSUB 1000
28 Y=7
29 M$="PULSATIONS PAR MINUTE. "
30 GOSUB 1000
31 PLOT 32,8,8
32 POKE 616,8
33 PRINT
34 POKE 617,34
35 INPUT P1
37 REM
38 REM * 2eme phase *
39 REM
40 Y=11
41 M$="EFFECTUEZ MAINTENANT 30 "
42 GOSUB 1000
43 Y=13
44 M$="FLEXIONS PUIS TAPEZ VOTRE "
```

```

45 GOSUB 1000
46 Y=15
47 M$="NOUVEAU POULS.           "
48 GOSUB 1000
50 PLOT 32,16,8
52 POKE 616,16
53 PRINT
54 POKE 617,34
55 INPUT P2
56 REM
57 REM * 3eme phase *
58 REM
60 Y=19
61 M$="REPOSEZ-VOUS PENDANT UNE  "
62 GOSUB 1000
63 Y=21
64 M$="MINUTE PUIS TAPEZ VOTRE  "
65 GOSUB 1000
66 Y=23
67 M$="NOUVEAU POULS.           "
68 GOSUB 1000
70 PLOT 32,24,8
72 POKE 616,24
73 PRINT
74 POKE 617,34
75 INPUT P3
76 REM
77 REM ** RESULTATS DU TEST **
78 REM
80 I=((P2-70)+2*(P3-P1))/10
90 I=INT(I/3)+1
150 CLS
160 Y=7:FD=20:AV=3
161 M$="AU REPOS                 "+STR$(P1)
162 GOSUB 1000
170 Y=11
171 M$="JUSTE APRES UN EFFORT "+STR$(P2)
172 GOSUB 1000
180 Y=15
181 M$="1 mn APRES L'EFFORT     "+STR$(P3)
182 GOSUB 1000

```

```

185 Y=21
187 IF I<1 OR I>4 THEN 270
190 RESTORE
200 FOR J=1 TO I
210 READ A$
220 NEXT J
250 M$="VOTRE FORME EST "+CHR$(14)+A$
260 GOSUB 1000
265 END
270 M$="VOUS ETES HORS NORMES ! "
275 ZAP:ZAP:ZAP
277 GOSUB 1000
280 Y=23
285 M$="CONSULTEZ VOTRE MEDECIN."
287 GOSUB 1000
290 END
300 DATA BONNE,MOYENNE,PASSABLE,MAUVAISE
985 END
989 REM
990 REM ** AFFICHAGE **
991 REM
1000 N$=CHR$(AT)+CHR$(FD)+CHR$(AV)
1010 FOR NN=1 TO NB
1020 PLOT X,Y+NN-1,N$+M$+CHR$(FL)
1030 NEXT NN
1040 RETURN

```

23. Touche sensible

Nombre de personnes : 1 à 80 (mais exclusivement des hommes majeurs et irresponsables).


Présentation :

Une seule parmi les lettres déclenche l'exécution d'une mélodie connue de tous.

A vous de presser la bonne touche.

L'ordinateuse est bonne princesse et elle vous donnera une nouvelle chance tant que vous ne l'aurez pas sollicitée inconvenablement.

Des versions de ce programme circulent sur d'autres ordinateurs ; cela prouve bien que le vice est universellement répandu dans cette nouvelle génération de machines.



Remarques :

L'affichage des 7 premiers segments du message final (lignes 210 à 270) est réalisé à l'aide de PLOT qu'il a fallu "tabuler" pour obtenir une phrase. Le paramètre de tabulation L doit donc être calculé 7 fois de suite et c'est le rôle de la ligne 220.

Programme :

```
1 REM
2 REM
3 REM
4 REM ** LA TOUCHE SENSIBLE **
5 REM -----
```

```

6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
20 REM
30 REM * MUSIQUE *
40 REM
50 A$(1)="48484846454545"
60 A$(2)="4445454545"
70 A$(3)="44454545"
80 A$(4)="44454848"
90 A$(5)="454846464343"
100 A$(6)="43434342434646454351515151"
110 A$(7)="41434548414345483841454343"
120 REM
130 REM * TEXTES *
140 REM
150 FOR I=7 TO 1 STEP -1
155 READ B$(I)
160 NEXT I
162 DATA .. ,MON,TOUCHER,OSE
165 DATA IL A,MAIS,OH !
167 REM
168 PRINT:PRINT:PRINT
169 M$=" TROUVEZ LA TOUCHE QUI FAIT"
170 PRINT CHR$(138);M$
171 PRINT CHR$(138);M$
172 PRINT:PRINT
173 M$=" JOUER UNE PETITE MUSIQUE."
174 PRINT CHR$(138);M$
175 PRINT CHR$(138);M$
177 PRINT:PRINT:PRINT
180 GET R$
182 PRINT R$;" ";
185 IF R$=CHR$(81) THEN ZAP:GOTO 210
190 SHOOT
195 GOTO 180
197 REM
198 REM * ACTION *

```

```

199 REM
210 L=1
215 FOR I=1 TO 7
220 L=L+LEN(B$(I-1))+1
225 PLOT 0,19,CHR$(14)+CHR$(1)
227 PLOT 0,20,CHR$(14)+CHR$(1)
230 PLOT L,19,B$(I)
240 PLOT L,20,B$(I)
250 M$=A$(I)
255 GOSUB 1000
270 NEXT I
274 PLOT L+3,19,"Q"
276 PLOT L+3,20,"Q"
280 MUSIC 1,4,1,0
290 MUSIC 2,4,5,0
300 MUSIC 3,4,8,0
310 PLAY 1,0,1,5000
320 PLOT 0,19,CHR$(10)
325 PLOT 0,20,CHR$(10)
900 END
1000 FOR J=1 TO LEN(M$)-1 STEP 2
1010 O=VAL(MID$(M$,J,1))
1020 N=VAL("#"+MID$(M$,J+1,1))
1030 MUSIC 1,O,N,0
1045 PLAY 1,0,1,5000
1050 WAIT 35
1060 NEXT J
1065 PLAY 0,0,0,0
1070 RETURN

```


24. Z ... mégalettres ... A

Nombre de personnes : 1

Présentation :

Voilà un mot qu'on vient de taper de façon ordinaire sur le clavier ; ce mot s'inscrit alors en grand sur l'écran sous forme de grosses lettres, elles-mêmes constituées du caractère en question ; voilà qui peut paraître magique !

Désolé de rompre le charme mais ce n'est là qu'une exploitation des codes binaires qui constituent le générateur de caractères.




Remarques :

Le programme comporte des points techniques délicats et difficile à expliquer en quelques phrases : essayons quand même de lever le voile.

Le cœur du programme, c'est la boucle FOR...NEXT qui s'étend de la ligne 75 à la ligne 200.


Pour comprendre il faut savoir que, sur l'écran, chaque caractère occupe un rectangle (on dit aussi une matrice) de 6 points de large sur 8 point de haut.




Exemple : la lettre A

Afin que l'ordinateur puisse le dessiner sur l'écran, les formes des caractères sont stockées en mémoire à partir de l'adresse 46080.

Les huit lignes de chaque matrice (considérées du haut vers le bas) occupent des mémoires consécutives, le caractère n° 0 est donc mémorisé de 46080 à 46087, le caractère n° 1 de 46088 à 46095, le caractère n° 65 (c'est la lettre A) de 46600 à 46607, et ceci de la façon suivante :




ou plutôt,



Car l'ordinateur ne connaît pas les blancs et les noirs mais les 0 et les 1 c'est-à-dire les chiffres binaires qu'on appelle "bits" en français.

En réalité, chaque "paquet" de 6 bits est précédé de deux 0 destinés à normaliser la taille des "paquets" à 8 afin d'obtenir des "octets".

On a donc, pour la lettre A :



Programme :

La formule de la ligne 100 calcule l'adresse du premier des huit octets correspondant à un caractère.

Exemple :

Si C\$ est la lettre A alors ASC(C\$), c'est-à-dire le code de A, vaut 65 (voir annexe), le calcul de C = 46080 + 65*8 donne bien 46600 comme indiqué plus haut.

La boucle 110 à 190 est chargée de prendre en considération, un à un, les 8 octets localisés par la formule précédente.

Quant à la boucle 130 à 180 imbriquée dans les deux précédentes, elle détecte, pour chacun des 6 points d'une rangée s'il est allumé (1) ou éteint (0). Pour cela on utilise la technique dite de "masquage" en effectuant un AND booleen avec l'une des six puissances de 2 contenues dans le tableau P et dont les représentations binaires sont :

pour 2⁰ : 000001
pour 2¹ : 000010
pour 2² : 000100

pour 2³ : 001000
pour 2⁴ : 010000
pour 2⁵ : 100000

Si le résultat du AND est 1 c'est que le point doit être allumé et on écrit alors, en vidéo-inverse'', et à l'endroit convenable de l'écran, le caractère qu'on est en train d'agrandir.

Je tenais à proposer ce programme et à vous fournir ces explications car la zone où figurent les formes des caractères est en mémoire vive et vous pourrez, si vous avez compris le principe de codage et de localisation du caractère, créer vos propres formes à l'aide de POKEs bien placés (voir le manuel de programmation en français fourni avec ORIC pages 98 et 99).

Programme :

```
1 REM
2 REM
3 REM
4 REM ** MEGALETTERS **
5 REM -----
6 REM
7 REM
8 REM
10 CLS
12 PAPER 6
15 INK 4
20 Y=12
22 REM
23 REM * DOUBLE HAUTEUR FIXE *
24 REM * MAGENTA SUR FOND JAUNE *
25 REM
27 CH$=CHR$(138)+CHR$(147)+CHR$(133)
28 PRINT
29 M$=" TAPEZ UN MOT (6 lettres maximum)"
30 PRINT CH$;M$
31 PRINT CH$;M$
32 PRINT:PRINT:PRINT
33 INPUT A$
34 A$=LEFT$(A$,6)
35 POKE 616,25
37 REM
```

```

38 REM * INHIBE LA SCRUTATION DU CLAVIER *
39 REM
40 CALLNE6DA
43 L=LEN(A$)
45 FOR I=10 TO 20
47 PLOT 0,I,CHR$(7)+CHR$(21)
48 PLOT 6*L+2,I,22
50 NEXT I
53 REM
55 REM ** SCRUTATION DU MOT **
60 REM
61 REM * 5 PREMIERES PUISSANCES DE 2 *
62 REM
63 FOR K=0 TO 5
65 P(K)=2^K
67 NEXT K
69 REM
70 REM * POUR CHAQUE LETTRE DU MOT *
71 REM
75 FOR I=1 TO L
80 X=6*(I-1)
90 C$=MID$(A$,I,1)
100 C=46080+ASC(C$)*8
104 REM
105 REM * POUR CHAQUE LIGNE DE LA MATRICE *
106 REM
110 FOR J=0 TO 7
120 OC=PEEK(C+J)
124 REM
125 REM * POUR CHAQUE OCTET D'UNE LIGNE *
126 REM
130 FOR K=5 TO 0 STEP -1
135 G=7-K
136 REM
137 REM * TEST D'UN BIT DE L'OCTET *
140 BI=OC AND P(K)
160 IF BI=0 THEN 180
170 PLOT X+G,Y+J,ASC(C$)+128
180 NEXT K
190 NEXT J
200 NEXT I

```

205 REM

206 REM * RETABLIT LA SCRUTATION-CLAVIER *

207 REM

210 CALL#E804

Annexes

ANNEXE 1

Code ASCII

Code	Caractère	Code	Caractère	Code	Caractère
032	ESPACE	064	Ⓐ	096	©
033	!	065	A	097	a
034	“	066	B	098	b
035	#	067	C	099	c
036	\$	068	D	100	d
037	%	069	E	101	e
038	&	070	F	102	f
039	,	071	G	103	g
040	(072	H	104	h
041)	073	I	105	i
042	*	074	J	106	j
043	+	075	K	107	k
044	,	076	L	108	l
045	-	077	M	109	m
046	/	078	N	110	n
047	.	079	O	111	o
048	0	080	P	112	p
049	1	081	Q	113	q
050	2	082	R	114	r
051	3	083	S	115	s
052	4	084	T	116	t
053	5	085	U	117	u
054	6	086	V	118	v
055	7	087	W	119	w
056	8	088	X	120	x
057	9	089	Y	121	y
058	:	090	Z	122	z
059	;	091	[123	{
060	<	092	\	124	
061	=	093]	125	}
062	>	094	↑	126	DEL
063	?	095	£	127	■


Les codes ASCII sont exprimés en décimal

ANNEXE 2

Les attributs

Sur l'écran, chaque caractère occupe un rectangle de 6 points de large sur 10 points de haut.

Exemple :



Deux informations sont nécessaires pour définir l'aspect coloré d'un caractère sur l'écran :

- la couleur des points qui dessinent sa forme
- la couleur du fond

Ces couleurs sont déterminées par des codes qui doivent figurer sur la même ligne d'écran que le caractère et à sa gauche.

Chaque code ainsi écrit occupe lui-même un rectangle 6 × 10 qui apparaît à l'écran sous la forme d'un pavé uniformément coloré.

Exemple :

Portion d'une
ligne d'écran



MESSAGE apparaît alors en rouge sur fond vert et IMPORTANT s'inscrit en noir sur fond vert également car seule la couleur de forme a été rédefinie entre les deux mots.

Les emplacements contenant les trois codes ainsi que la fin de la ligne sont, ici, en vert.

Pour obtenir cet effet il faut, bien entendu, connaître les codes qui conviennent et qui font l'objet d'un tableau proposé plus loin.

En mode TEXT, il y a trois façon de positionner ces codes.

1 — par PRINT

```
10 PRINT CHR$(146);CHR$(129);"MESSAGE";  
CHR$(128);"IMPORTANT"
```

provoquera l'effet évoqué plus haut.

On pourrait d'ailleurs remplacer les ; par des + et écrire :

```
10 PRINT CHR$(146)+CHR$(129)+"MESSAGE"+  
CHR$(128)+"IMPORTANT"
```

2 — par PLOT

```
10 PLOT 3,17,CHR$(18)+CHR$(1)+"MESSAGE"+  
CHR$(0)+"IMPORTANT"
```

provoquera le même effet.

Ici on est obligé d'utiliser le signe + de concaténation car le 3^e argument de PLOT est une chaîne.

Les arguments 3 et 17 permettent simplement d'écrire à partir de la 3^e case de la ligne 17 sur l'écran.

3 — par POKE

POKE est une instruction qui permet de modifier directement le contenu d'une case mémoire et qui doit, de ce fait, être utilisée avec précaution. L'écran étant la duplication d'une zone de la mémoire, il suffit d'écrire "le bon code au bon endroit !"

La zone de mémoire dédiée à l'écran comporte 1120 cases numérotée de 48 000 à 49 119 et dont les 40 premières sont réservées à la ligne de service située tout en haut de l'écran et sur laquelle s'inscrit CAPS.

Ceci étant dit, faites exécuter, le programme suivant dans le seul but d'écrire quelquechose sur l'écran.

```
10 CLS  
20 PAPER 6  
30 FOR I=1 TO 10  
40 PRINT "QWERTYUIOPASDFGHJKL"  
50 NEXT I
```

puis tapez, sans numéro de ligne :

```
POKE 48087,1 return
```

les lettres UIOPASDFGHJK de la 2^e ligne sont maintenant rouges et le caractère Y a disparu ou plutôt il a été remplacé par le code de "forme rouge"

Tapez maintenant

POKE 48086,19 `return`

vous obtenez une partie de la 2^e ligne sur fond jaune et la lettre T a disparu au profit du code de "fond jaune".

Enfin tapez

POKE 48101,22 `return`

cela rétablit le fond bleu sur la fin de la ligne

Remarque :

il n'est pas possible d'écrire les 4 premières lettres d'un mot d'une couleur et le reste du mot d'une autre couleur sauf à accepter de séparer les deux portions par un espace.

Voici maintenant le tableau où figurent les codes correspondant aux couleurs de forme et de fond.

	FORME		FOND	
	par PRINT	par PLOT ou POKE	par PRINT	par PLOT ou POKE
noir	128	0	144	16
rouge	129	1	145	17
vert	130	2	146	18
jaune	131	3	147	19
bleu	132	4	148	20
magenta	133	5	149	21
cyan	134	6	150	22
blanc	135	7	151	23

Quand on utilise PRINT ou PLOT, le code doit être mis en argument de CHR\$.

Exemple :

PRINT CHR\$ (147) ou PLOT 5,14,CHR\$ (19)

pour obtenir un fond jaune

Quittons la couleur pour aborder les deux autres attributs que ORIC 1 permet de définir :

- caractère fixe ou clignotant
- caractère en simple ou double hauteur.

Le mode de transmission de code est également triple (PRINT, PLOT et POKE) et fonctionne exactement de la même manière que pour les attributs de couleur.

		par PRINT	par PLOT ou POKE
simple hauteur	fixe	136	8
double hauteur	fixe	138	10
simple hauteur	clignotant	140	12
double hauteur	clignotant	142	14

Exemple :

```
10 PRINT CHR$(146);CHR$(129);"MESSAGE";  
CHR$(140);"IMPORTANT"
```

ou

```
10 PLOT 3,17,CHR$(18)+CHR$(1)+"MESSAGE"+  
CHR$(12)+"IMPORTANT"
```

ATTENTION

Essayez maintenant

```
10 CLS  
20 PLOT 3,17,CHR$(10)+"IMPAIR"
```

Seule la partie supérieure des lettres double hauteur apparaît !

Ajoutez la ligne 20 pour obtenir

```
10 CLS  
20 PLOT 3,17,CHR$(10)+"IMPAIR"  
30 PLOT 3,18,CHR$(10)+"IMPAIR"
```

l'affichage est complet ; maintenant modifiez les arguments de PLOT correspondant aux lignes d'écran afin d'obtenir

```
10 CLS
20 PLOT 3,19,CHR$(10)+"IMPAIR"
30 PLOT 3,19,CHR$(10)+"IMPAIR"
```

la partie supérieure des lettres apparaît... sous la partie inférieure !

Moralité : en double hauteur la partie supérieure doit toujours être écrite sur une ligne d'écran de numéro impair.

Cela pose un problème lorsqu'on veut écrire en double hauteur avec des PRINT car on ne sait pas toujours si la ligne d'écran sera de numéro pair ou impair.

Exemple :

```
10 CLS
20 PRINT
30 PRINT CHR$(138);"MEFIANCE"
40 PRINT CHR$(138);"MEFIANCE"
50 PRINT
60 PRINT CHR$(138);"MEFIANCE"
70 PRINT CHR$(138);"MEFIANCE"
```

la premier MEFIANCE est bon, le second est mauvais.

Supprimez la ligne 20.

Le premier est maintenant mauvais et le second est bon.

Ajoutez la ligne

```
55 PRINT
```

Les deux sont mauvais.

Supprimez les lignes 50 et 55 et rajoutez la ligne.

Victoire, les deux sont bons !

moralité : MEFIANCE !


L'impression de ce livre
a été réalisée sur les presses
des Imprimeries Aubin
à Poitiers/Ligugé



pour les Éditions Cedic

Achévé d'imprimer en janvier 1984
N° d'impression, L 16191
Dépôt légal, janvier 1984

Imprimé en France



"LE HERON"
864
578 842 518 893 173 296 ↑ 993 54
choisissez-vous le nombre fleche
(O ou N)
RATE 1 LE PLUS GRAND ETAIT 993
PUSSEZ 'C' POUR RECOMMENCER
OU 'R' POUR PESSER CE EU

ORIC
ORIC

TAPES VOUS TAPPEZ
? 93 12 65
XEZAML XFZCML WEZANJ XDZCMK
NFZANK MFZBML MFZCMK XDZAMK
WEZBML HDZCML XFZCMK HDZBML
XEZBML YEZCML XDZBNJ YFZCML
YFZBNK HDZBML WEZBNJ HFZANJ
XDZBNK

Ready
LE COMPTE EST BON
ESSAI NUMERO 10
BRAVO

AU REPOS
JUSTE APRES UN EFFORT
UN MOMENT APRES L'EFFORT
VOTRE FORME EST MOYENNE

CALCUL DE VOTRE RATION CALORIQUE
VOTRE POIDS en Kg 87
VOTRE TAILLE en cm 172
VOTRE TAUX en cm² 179
VOTRE SEXE (M ou F) M
VOTRE ACTIVITE (0 à 29) 12
VOS BESOINS ENERGETIQUES
2705 CALORIES PAR JOUR

TAQUIN
A O C B W
F H G J
I E P X N
K L N S T
R D O U U



Collection dirigée par **ANDRÉ DELEDICQ**

L'ouvrage

Pour vous divertir en faisant plus ample connaissance avec votre ORIC, voici vingt-quatre programmes originaux commentés. Ils ont tous été testés et listés sur l'ORIC 16 K. Bien sûr ils peuvent tous fonctionner sur l'ORIC 48 K ou plus. A vous de jouer !

CEDIC/FERNAND NATHAN

ISBN 2-7124-0539-0

59 F